code:
;-------------------------------------------------
;约会だ！约会に行くぞ！！
;-------------------------------------------------
@EVENT_DATE_MAIN
#DIM MEMO_REALTIMEFORFLAG
#DIM LCOUNT
;換装前の衣装の記録
#DIMS MEMO_CLO, 50
#DIM MEMO_EQUIP, 50
#DIM MEMO_STAIN, 20
#DIMS MEMO_PLACE

SIF CHARA_NUM("约会可能") == 0
	RETURN 0
;とりあえずは周日の昼のみ
SIF DAYEV("约会領帯ム") == 0
	RETURN 0

MEMO_PLACE = %SAVESTR:現在位置%

MEMO_REALTIMEFORFLAG = FLAG:現実時間
;現実時間フラグ取得
CALL GET_REALTIMEFORFLAG

REDRAW 0

;約束があるならば既に相手は決まっている
IF COND("约会の予定有")
	TARGET = TO_NUMBER(CFLAG:MASTER:约会的约定)
	PRINTFORMW %CALLNAME:MASTER%去了和%NAMES("约会相手")%的约定好见面的地方。
;約束していないなら相手選び
ELSE
	CALL SELECT_DATE
	;やっぱやーめた
	IF RESULT == 0
		PRINTFORMW 不去约会了。
		FLAG:現実時間 = MEMO_REALTIMEFORFLAG
		RETURN 0
	ENDIF

	CVARSET CFLAG, GETNUM(CFLAG, "睡眠")
	CVARSET CFLAG, GETNUM(CFLAG, "刚睡醒")

	CALL BLANKLINE
	;まずは誘った際の反応
	CALL KOJO_DATE, "邀请去约会"

	IF TFLAG:地の文スキップ == 0
		;この場に居ない相手を選んだのなら、呼び出したりなんなり
		IF COND("同居中")
			PRINTFORMW 难得地有了一次约会、%CALLNAME:MASTER%他们决定各自做好准备后在門外汇合然后出发。
		ELSE
			PRINTFORMW 成功邀请了%CALLNAME:TARGET%的%CALLNAME:MASTER%、为了享受约会决定在外面汇合。
		ENDIF
	ENDIF
ENDIF

;约会中フラグON
SETBIT FLAG:状況判定, 40
FLAG:情緒 = 30

;约会の行き先初期化
SAVESTR:约会 =
STR:Ｈする場所 =

CFLAG:睡眠 = 0
CFLAG:刚睡醒 = 0
;胖次などのCFLAGは謎にする
CVARSET CFLAG, GETNUM(CFLAG, "内衣確認済")
CVARSET CFLAG, GETNUM(CFLAG, "股間確認済")
CVARSET CFLAG, GETNUM(CFLAG, "检查阴部")
CVARSET CFLAG, GETNUM(CFLAG, "掀起裙子")

;约会で使うTFLAG初期化
TFLAG:牵手 = 0
TFLAG:接吻 = 0
TFLAG:手淫 = 0
TFLAG:口交 = 0
TFLAG:摸屁股 = 0
TFLAG:性交 = 0
TFLAG:约会中牵手可能 = 0
TFLAG:约会中H可能 = 0
TFLAG:偷瞄内衣可能 = 0
TFLAG:约会撮影可能 = 0
TFLAG:泳装约会 = 0
TFLAG:文化祭约会 = 0
TFLAG:衣装解説済み = 0
TFLAG:事件回数 = 0

CALL EVENT_DATE, "约会衣装に更衣"
DRAWLINE
DRAWLINE
;约会プラン羅列
CALL EVENT_DATE, "待ち合わせ"

;着ていた服と汚れを記憶する
FOR LCOUNT, 40, 50
	MEMO_CLO:LCOUNT = %CSTR:LCOUNT%
NEXT
FOR LCOUNT, 40, 50
	MEMO_EQUIP:LCOUNT = EQUIP:LCOUNT
NEXT
FOR LCOUNT, 10, 20
	MEMO_STAIN:LCOUNT = STAIN:LCOUNT
NEXT

CALL EVENT_DATE, "行先決定"
CALL EVENT_DATE, "内衣チラ見え"

;ここから先は行き先次第で変化がある
SELECTCASE SAVESTR:约会
CASE "文化祭"
	;制服に更衣
	FOR LCOUNT, 2, CHARANUM
		SIF LCOUNT == TARGET
			CONTINUE
		SIF COND("学園在籍", LCOUNT)
			CALL SET_COORDINATE, "制服", LCOUNT
	NEXT
	;文化祭のCEVENTフラグ重置
	FOR LCOUNT, 1, CHARANUM
		CALL DEL_CEVENT_GROUP, "学園祭", LCOUNT
	NEXT
	CALL EVENT_DATE, "文化祭"
	CALL EVENT_DATE, "便当"
	CALL EVENT_DATE, "牵手"
	CALL EVENT_DATE, "文化祭"
	CALL EVENT_DATE, "约会撮影"
	CALL EVENT_DATE, "约会H"
	CALL EVENT_DATE, "文化祭"
CASE "去游泳"
	CALL EVENT_DATE, "着替えて合流"
	CALL EVENT_DATE, "便当"
	CALL EVENT_DATE, "约会撮影"
	CALL EVENT_DATE, "牵手"
	CALL EVENT_DATE, "约会H"
CASE "自驾游约会"
	CALL EVENT_DATE, "出発"
	CALL EVENT_DATE, "風景"
	CALL EVENT_DATE, "便当"
	CALL EVENT_DATE, "约会H"
CASE "去电影院"
	CALL EVENT_DATE, "牵手"
	CALL EVENT_DATE, "约会H"

	;この領帯ミングで映画館から出る
	SAVESTR:現在位置 = %MEMO_PLACE%

	CALL EVENT_DATE, "便当"
;2022/12/16 hokaguru 追記
CASE "在网咖度过"
	CALL EVENT_DATE, "牵手"
	CALL EVENT_DATE, "约会H"
CASE "去图书馆"
	CALL EVENT_DATE, "便当"
	CALL BLANK_DRAWLINE
	CALL SETFLAG_DAILY, "勉強"
	CALL EVENT_DATE, "约会H"
CASE "尋訪美食"
	CALL EVENT_DATE, "牵手"
	;泳装を買いに行った場合など
	CALL EVENT_DATE, "购物"
	CALL EVENT_DATE, "约会撮影"
	CALL EVENT_DATE, "约会H"
;2021/07/27　玛丽酱の中の人が書いた
CASE "温泉旅行"
	CALL EVENT_DATE, "牵手"
	CALL EVENT_DATE, "约会撮影"
	CALL EVENT_DATE, "约会H"
CASE "访问教会"
	CALL EVENT_DATE, "教会午餐"
	CALL EVENT_DATE, "牵手"
	IF TALENT:淫乱 && TALENT:恋慕
		CALL EVENT_DATE, "特别忏悔室"
	ENDIF
	CALL EVENT_DATE, "约会エロ"
;お家约会。現状乃々君限定だが、他角色にも流用できるように
;2023/12/02追記
CASE @"%CALLNAME:TARGET%的家"
	SELECTCASE CALLNAME:TARGET
	CASE "乃々"
		CALL EVENT_DATE, "约会撮影"
		CALL EVENT_DATE, "约会H"
	CASEELSE
		CALL EVENT_DATE, "便当"
		CALL EVENT_DATE, "牵手"
		CALL EVENT_DATE, "约会撮影"
		CALL EVENT_DATE, "约会H"
	ENDSELECT
CASEELSE
	CALL EVENT_DATE, "便当"
	CALL EVENT_DATE, "牵手"
	;泳装を買いに行った場合など
	CALL EVENT_DATE, "购物"
	CALL EVENT_DATE, "约会撮影"
	CALL EVENT_DATE, "约会H"
ENDSELECT
;泳装约会は晒痕も起きる
SIF TFLAG:泳装约会
	CALL EVENT_DATE, "晒痕"

;着ていた服と汚れを戻す
FOR LCOUNT, 40, 50
	CSTR:LCOUNT = %MEMO_CLO:LCOUNT%
NEXT
FOR LCOUNT, 40, 50
	EQUIP:LCOUNT = MEMO_EQUIP:LCOUNT
NEXT
FOR LCOUNT, 10, 20
	STAIN:LCOUNT = MEMO_STAIN:LCOUNT
NEXT

;時間帯を傍晚に変更
TIME = 3

;约会も終わり。さて……
CALL BLANK_DRAWLINE
IF TFLAG:文化祭约会
	PRINTFORMW 注意到大多数展区已经開始在做撤収的準備了、文化祭也临近尾声了。
ELSE
	PRINTFORMW 愉快的约会、差不多要到结束的时间了。
ENDIF
CALL BLANKLINE
CALL KOJO_DATE, "帰り際"

SAVESTR:現在位置 = %MEMO_PLACE%

RETURN 1

;-------------------------------------------------
;约会のお相手を選ぶ
;-------------------------------------------------
@SELECT_DATE
#DIM CHOICE
#DIM MEMO_LINECOUNT
MEMO_LINECOUNT = LINECOUNT

REDRAW 0

PRINTL ……要邀请谁一起约会呢？
CALL PRINT_STRL, "顺便一提、需要_黄色_[信頼LV3]or[恋慕]or[屈服刻印LV1]_才会接受邀请"
CALL SHOW_CHARA_LIST, "约会"

$INPUT_LOOP
INPUT

CHOICE = RESULT

SELECTCASE CHOICE
CASE 100
	RETURN 0
CASE IS < MASTER, IS >= CHARANUM
	GOTO INPUT_LOOP
CASEELSE
	SIF COND("约会可能", CHOICE) == 0
		GOTO INPUT_LOOP
ENDSELECT

;この場に居ない相手を選んだのなら、優先順位は変動させない
IF COND("現在不在", CHOICE)
	TARGET = CHOICE
ELSE
	CALL SETFLAG, "対象指定", CHOICE
ENDIF

RETURN 1

;-------------------------------------------------
;约会のイベント管理
;-------------------------------------------------
@EVENT_DATE, ARGS
;起きうるイベントの判定に使う
#DIMS EVENT_TABLE
#DIMS CHOICES
#DIM ACTOR
#DIM LCOUNT
#DIM MEMO_LINECOUNT
#DIM MEMO_MOOD
#DIM MEMO_EJECT
;換装前の衣装の記録
#DIMS BEFORE_CLO
#DIMS AFTER_CLO

TFLAG:地の文スキップ = 0
MEMO_MOOD = FLAG:情緒

SELECTCASE ARGS
CASE "约会衣装に更衣"
	;约会用の髪型など変更
	CALL SETFLAG, "ＴＰＯによる外見変更"

	;振袖の時は着替えない
	IF CHECK_SHIRT("振袖")
		FLAG:情緒 += 30
	ELSE
		STR:変更衣装 = 约会衣装

		;着ていた服を記憶する。一致するかを見るだけなので文字列コピーで良い
		BEFORE_CLO = %CSTR:胖次%%CSTR:胸罩%%CSTR:服%%CSTR:裙%%CSTR:贴身制服%%CSTR:外套%%CSTR:襪子%%CSTR:靴%%CSTR:手套%%CSTR:飾品%

		CALL KOJO_DATE, "约会衣装に更衣"

		AFTER_CLO = %CSTR:胖次%%CSTR:胸罩%%CSTR:服%%CSTR:裙%%CSTR:贴身制服%%CSTR:外套%%CSTR:襪子%%CSTR:靴%%CSTR:手套%%CSTR:飾品%

		;着替えた場合はフラグON
		SIF BEFORE_CLO != AFTER_CLO
			CALL SETFLAG, "着替えをした"

		;口上などで着替えていないなら(あるなら)约会衣装になる
		SIF COND("约会衣装に着替えた") == 0
			CALL SETFLAG, "约会衣装に更衣"
		;约会用の衣装で登場したなら情緒アップ
		SIF COND("约会衣装に着替えた")
			FLAG:情緒 += 30
	ENDIF
CASE "待ち合わせ"
	CALL EMOTION_PLUS, "甜言蜜语/对话框兴奋/布", TARGET
	CALL PRINT_CUTIN, "縮小", TARGET
	CALL KOJO_DATE, "待ち合わせ"
	;もしも衣装の解説が無いならここでしておく
	IF TFLAG:衣装解説済み == 0
		IF COND("约会衣装に着替えた") && TFLAG:地の文スキップ == 0
			CALL PRINT_STRW, @"今天的%CALLNAME:TARGET%穿的是、_爱心粉紅_%CLO_ABOUT(TARGET)%_。"
			PRINTFORM %TEXTR("和平时不同的/进入了和往常不同的气势的/看起来是为了约会而精心准备过的")%%CONDS("間柄")%的姿态、
			PRINTFORMW 情绪十分高涨……
		ELSE

		ENDIF
		TFLAG:衣装解説済み = 1
	ENDIF
CASE "行先決定"
	;相手が行きたい場所あったらそっち優先
	IF SAVESTR:约会 == ""
		;初めてのお约会！
		IF EXP:MASTER:约会経験 == 0
			PRINTFORML 值得纪念的第一次约会、该如何度过呢？
		;この相手とは初めて
		ELSEIF EXP:约会経験 == 0
			PRINTFORML 这次是和%CALLNAME:TARGET%的第一次约会。做些什么呢？
		ELSE
			PRINTFORML 这次的约会、以怎样的计划进行呢？
		ENDIF
		SIF COND("夏") && CSTR:今年の泳装 != "" && COND("今年の泳装選択済")
			CALL PRINT_STRL, @"（※为了游泳池和海准备的_黄色_今年的泳装(%SWIM(CSTR:今年の泳装, "種類")%)_已经購入済了）"

		;起きうるイベントの判定に使う
		STR:選択肢 =

		;口上側で選択肢を自由に作ることができる
		TRYCALLFORM DATE_K{NO:TARGET}_ADDSELECT

		SIF DAYEV("文化祭")
			STR:選択肢 += "文化祭/"

		SELECTCASE TEXTS("季節詳細")
		CASE "春", "初夏", "夏"
			SIF COND("今年の泳装選択済") == 0 && COND("池面") == 0
				STR:選択肢 += "去买泳衣/"
		ENDSELECT

		;季節イベント。特殊な行先は作りかけなので自重して封印
		IF COND("春")
	;		STR:選択肢 += "赏花/"
		ELSEIF COND("夏")
			STR:選択肢 += "去游泳/"
	;		STR:選択肢 += "爬山/"
		ELSEIF COND("秋")
	;		STR:選択肢 += "赏红叶/"
		ELSEIF COND("冬")
	;		STR:選択肢 += "滑雪/"
		ENDIF

		;2021/07/27　玛丽酱の中の人が書いた
		;もうちょいイベント充実したら実装予定
		;SIF TALENT:相思相愛
		;	STR:選択肢 += "温泉旅行/"

		;慣れた相手となら身近な場所を選べる
		SIF EXP:约会経験
			STR:選択肢 += "散歩/"

		;通年ＯＫなスポット
		STR:選択肢 += "尋訪美食/逛街/去水族馆/去游乐园/去电影院/在网咖度过/"

		;通年ＯＫな乗り物约会
		;教師と研究生限定にしてみる
		SIF STRCOUNTS(CSTR:学年, "研究生", "教師")
			STR:選択肢 += "自驾游约会/"

		;図書館は学生限定
		SIF STRCOUNTS(CSTR:学年, "年生", "初等部", "中等部")
			STR:選択肢 += "去图书馆/"

		VARSET LOCALS
		SPLIT STR:選択肢, "/", LOCALS
		CALL PRINT_SELECT, STR:選択肢
		SAVESTR:约会 = %LOCALS:RESULT%
	ENDIF

	;行先に応じて場所変更
	SELECTCASE SAVESTR:约会
	CASE "自驾游约会"
		SAVESTR:現在位置 = 自驾游
	CASE "文化祭"
		SAVESTR:現在位置 = 文化祭
	CASE "去买泳衣"
		SAVESTR:現在位置 = 泳装屋
	CASE "去游泳"
		SAVESTR:現在位置 = 泳池
	CASE "散歩"
		SAVESTR:現在位置 = %SAVESTR:约会%
	CASE "尋訪美食"
		SAVESTR:現在位置 = %SAVESTR:约会%
	CASE "逛街"
		SAVESTR:現在位置 = %SAVESTR:约会%
	CASE "去水族馆"
		SAVESTR:現在位置 = 水族館
	CASE "去游乐园"
		SAVESTR:現在位置 = 遊園地
	CASE "去电影院"
		SAVESTR:現在位置 = 映画館
	CASE "在网咖度过"
		SAVESTR:現在位置 = 网咖
	CASE "去图书馆"
		SAVESTR:現在位置 = 図書館
	ENDSELECT

	MEMO_LINECOUNT = LINECOUNT
	;口上で作った選択肢を選んだならここで地の文と台詞が出る
	TRYCALLFORM KOJO_DATE_K{NO:TARGET}_ADDSELECT

	;口上で作った選択肢を選んでいない場合
	SIF LINECOUNT == MEMO_LINECOUNT
		CALL KOJO_DATE, "约会本番"

	CALL BLANKLINE

	;情緒のBOUNS計算
	SIF TFLAG:牵手
		FLAG:情緒 += 15

	;约会場所によっていくつかのフラグが立つ
	;情緒BOUNSが高い場所でFLAG:情緒+60、普通ならFLAG:情緒+30、约会っぽくない場所だと修正无
	SELECTCASE SAVESTR:约会
	;高情緒＆エロ可能
	;2021/07/27　玛丽酱の中の人が書いた
;	CASE "去水族馆", "去游乐园"
	CASE "去水族馆", "去游乐园", "温泉旅行"
		FLAG:情緒 += 60
		TFLAG:约会中H可能 = 1
		TFLAG:约会中牵手可能 = 1
		TFLAG:偷瞄内衣可能 = 1
		TFLAG:约会撮影可能 = 1
	;2019/10 hokaguru 追記 映画館で内衣は見えへんやろう……
	CASE "去电影院"
		FLAG:情緒 += 60
		TFLAG:约会中H可能 = 1
		TFLAG:约会中牵手可能 = 1
	;2022/12/16 hokaguru 追記
	CASE "在网咖度过"
		FLAG:情緒 += 60
		TFLAG:约会中H可能 = 1
		TFLAG:约会中牵手可能 = 1
		TFLAG:偷瞄内衣可能 = 1
	CASE "去游泳"
		FLAG:情緒 += 30
		TFLAG:约会中H可能 = 1
		TFLAG:约会中牵手可能 = 1
		TFLAG:泳装约会 = 1
		TFLAG:约会撮影可能 = 1
	;並情緒＆エロ可能
	CASE "文化祭"
		FLAG:情緒 += 30
		TFLAG:文化祭约会 = 1
		TFLAG:约会中牵手可能 = 1
		TFLAG:约会中H可能 = 1
		TFLAG:偷瞄内衣可能 = 1
		TFLAG:约会撮影可能 = 1
	CASE "去买泳衣"
		FLAG:情緒 += 30
		TFLAG:约会中牵手可能 = 1
		TFLAG:约会中H可能 = 1
		TFLAG:偷瞄内衣可能 = 1
		TFLAG:约会撮影可能 = 1
	;低情緒だけどエロ可能
	CASE "去图书馆"
		FLAG:情緒 += 30
		TFLAG:约会中H可能 = 1
		TFLAG:偷瞄内衣可能 = 1
	CASE "散歩"
		TFLAG:约会中H可能 = 1
		TFLAG:约会中牵手可能 = 1
		TFLAG:偷瞄内衣可能 = 1
		TFLAG:约会撮影可能 = 1
	;自驾游约会。エロいことは出来るが流石に撮影をするほどの余裕はない
	CASE "自驾游约会"
		FLAG:情緒 += 60
		TFLAG:约会中H可能 = 1
		TFLAG:偷瞄内衣可能 = 1
	;其他。一般にエロいことは出来ない
	CASE "尋訪美食", "逛街"
		FLAG:情緒 += 30
		TFLAG:约会中牵手可能 = 1
		TFLAG:偷瞄内衣可能 = 1
		TFLAG:约会撮影可能 = 1
	;未設定
	CASE "赏花", "爬山", "赏红叶", "滑雪"
		FLAG:情緒 += 30
		TFLAG:约会中牵手可能 = 1
		TFLAG:偷瞄内衣可能 = 1
		TFLAG:约会撮影可能 = 1
	CASEELSE
		FLAG:情緒 += 30
		TFLAG:约会中牵手可能 = 1
		TFLAG:偷瞄内衣可能 = 1
		TFLAG:约会撮影可能 = 1
	ENDSELECT
CASE "内衣チラ見え"
	SIF TFLAG:偷瞄内衣可能 == 0
		RETURN 0
	CALL BLANK_DRAWLINE
	MEMO_LINECOUNT = LINECOUNT
	CALL EVENT_ACADEMY, "チラ見え"

	SIF LINECOUNT == MEMO_LINECOUNT
		CLEARLINE 1
CASE "着替えて合流"
	SELECTCASE SAVESTR:约会
	CASE "去游泳"
		SIF COND("今年の泳装選択済")
			SAVESTR:現在位置 = 海
		SIF CSTR:今年の泳装 == ""
			CALL EVENT_SHOP_SWIM, "自動選択"
		CALL SET_COORDINATE, "今年の泳装"
		CALL BLANK_DRAWLINE
		PRINTFORMW %CALLNAME:MASTER%一行人到达了目的地、决定暂时分開、在更衣室内换完衣服再汇合。
		PRINTW ………
		PRINTW ……
		PRINTW …
		IF TALENT:害羞 && TALENT:恋慕 == 0
			CALL EMOTION_PLUS, "害羞"
		ELSE
			CALL EMOTION_PLUS, "小悪魔/布"
		ENDIF
		SIF COMMENT_SWIM(CSTR:今年の泳装, "単独文") == ""
			CALL PRINT_CUTIN, "見たまま前後", TARGET
		CALL KOJO_DATE, "泳装に着替えた"
		PRINTFORMW 几分钟之后、%CALLNAME:MASTER%见到了换好%SWIM(CSTR:今年の泳装)%的%CALLNAME:TARGET%的身姿！
		IF COMMENT_SWIM(CSTR:今年の泳装, "単独文") != ""
			CALL PRINT_CUTIN, "見たまま前後", TARGET
			PRINTFORMW %COMMENT_SWIM(CSTR:今年の泳装, "単独文")%
		ENDIF
		FLAG:情緒 += 30
	ENDSELECT
CASE "出発"
	PRINTFORMW 到达了约定地点的%CALLNAME:TARGET%们開始商量起了関于目的地的话题
	PRINTFORMW 在听了各种各样的意见之后、%CALLNAME:TARGET%决定――
	CALL PRINT_SELECT, "去海边/去山里/去逛街"
	SELECTCASE RESULT
	CASE 0
		PRINTFORMW 决定了去附近的海边看看
		SAVESTR:現在位置 = 海边
	CASE 1
		PRINTFORMW 决定去附近的山里
		SAVESTR:現在位置 = 山道
	CASE 2
		PRINTFORMW 决定一起去逛街
		SAVESTR:現在位置 = 街中
	ENDSELECT
CASE "風景"
	PRINTFORMW %CALLNAME:MASTER%眺望着\@SAVESTR:現在位置 == "街中" ? 见惯了的街道的 # 前往郊外的电车外连绵不绝的自然\@风景
	PRINTFORM 从打開的窗边
	IF COND("夏")
		PRINTFORM 凉爽又清新的
	ELSEIF COND("冬")
		PRINTFORM 隆冬腊月的寒
	ELSE
		PRINTFORM 舒服的
	ENDIF
	PRINTFORMW 风吹了进来
	SELECTCASE CEVENTS("場所")
	CASE "街中"
		SELECTCASE TEXTS("季節詳細")
		CASE "冬"
			PRINTFORM 深蓝的天空与侵肌的寒冷
		CASE "夏"
			PRINTFORM 令人頭晕目眩的暑气
		CASE "秋"
			PRINTFORM 轻柔的秋风与步道两边盛開的秋菊
		CASE "春"
			PRINTFORM 蒙上了淡色的天空与甜蜜的花香
		ENDSELECT
		PRINTFORMW 如同在不认识的城镇中行走一般让%CALLNAME:MASTER%们产生了别样的新鲜感
	CASE "海边"
		SELECTCASE TEXTS("季節詳細")
		CASE "冬"
			PRINTFORMW 空寂的%TEXTR("被云层覆盖的天空/雪白的天空/青空")%与时刻不停地拍打在海岸上的海浪、一同构成了一幅凄美的画面
		CASE "夏"
			PRINTFORMW 璀璨地照耀着大地的太阳与海边游人们欢乐的笑声、让%CALLNAME:MASTER%的内心也不由得也一起欢呼雀跃起来
		CASE "秋"
			PRINTFORMW 被鲜艳的红叶与深蓝色的清空染上了颜色的椰树边、%CALLNAME:MASTER%的心也一同放空融入进了这美景中
		CASE "春"
			PRINTFORMW 望向那些向着海边飞散的樱花、%CALLNAME:MASTER%不由得開始质疑自己是否是闯入了画中世界
		ENDSELECT
	CASEELSE
		SELECTCASE TEXTS("季節詳細")
		CASE "冬"
			PRINTFORMW PRINTFORMW 空与大地的边界模糊不清、黑色的树木穿插其中的景象令人震撼、微微寒风刺骨、%CALLNAME:MASTER%不禁颤抖起来。
		CASE "夏"
			PRINTFORMW 从茂盛的树丛间漏下的阳光与周围不绝的蝉鸣一同营造出了轻松的氛围
		CASE "秋"
			PRINTFORMW 逐渐染上了鲜红的叶片与越发凉爽的秋风、一同宣告了后半年的到来
		CASE "春"
			PRINTFORMW 鲜艳地怒放的樱花在树枝上狂乱的点缀、就像是在欢迎%CALLNAME:TARGET%们的到来一般摇晃着
		ENDSELECT
	ENDSELECT
CASE "便当"
	SIF CEVENT("约会の便当") == 0
		RETURN 0
	CALL BLANK_DRAWLINE
	IF RAND:4 == 0
		SELECTCASE SAVESTR:约会
		CASE "自驾游约会"
			PRINTFORM 一边眺望着这美景一边闲谈着、话题自然转到了食物上
		CASE "去电影院"
			PRINTFORM 和%CALLNAME:TARGET%闲聊的同时走出了电影院、话题自然地转移到了食物上面
		CASEELSE
			PRINTFORM 移动的同时和%CALLNAME:TARGET%闲聊着、话题自然地转移到了食物上面
		ENDSELECT
	ELSE
		PRINTFORM 突然间%CALLNAME:MASTER%的肚子%TEXTR("发出了野兽般的吼声/咕～咕噜～地响了起来/发出了代表饥饿的声响")%、%CALLNAME:TARGET%
		IF TALENT:坦率 + TALENT:開朗 && TALENT:大力士 + TALENT:粗野 + TALENT:笨蛋
			PRINTFORM %TEXTR("噗的一声笑了出来/愉快地调侃了起来/咧嘴笑了起来")%
		ELSEIF TALENT:坦率 + TALENT:開朗 + TALENT:温柔 + TALENT:献身 + TALENT:冷静
			PRINTFORM %TEXTR("露出了笑容/温柔地微笑着/笑着靠过身来")%
		ELSEIF TALENT:強気 + TALENT:高慢
			PRINTFORM %TEXTR("忍俊不禁地笑出了声/笑着摆出\"不过如此啊\"的手势/用手肘戳着肋部偷偷地笑着")%
		ELSE
			PRINTFORM %TEXTR("笑出声来了/微微一笑/露出了略带惊讶的笑容")%
		ENDIF
	ENDIF
	PRINTFORMW ……差不多是午餐的时间了。
	PRINTL
	IF COND("夏")
		PRINTFORM %CALLNAME:MASTER%将%CALLNAME:TARGET%从保温箱中取出的便当、
	ELSE
		PRINTFORM %CALLNAME:MASTER%将%CALLNAME:TARGET%亲手用心做的便当、
	ENDIF
	IF COND("冬")
		PRINTFORMW 在早就计划好的室内广场的座位上一起享用了。
	ELSE
		SELECTCASE SAVESTR:约会
		CASE "去水族馆", "去游乐园", "去电影院", "滑雪", "去游泳", "去买泳衣"
			PRINTFORMW 在长凳上一起享用了。
		CASE "去图书馆"
			PRINTFORMW 在馆外的长凳上一起享用了。
		CASE "散歩", "尋訪美食", "逛街"
			PRINTFORMW %TEXTR("在公園的长凳上/在咖啡館的開放空间里/在购物中心的開放空间里")%一起享用了。
		CASE "赏花", "爬山", "赏红叶"
			PRINTFORMW 坐在休闲座椅上一起享用了。
		CASE "自驾游约会"
			PRINTFORMW 在车中一起吃饭。
		CASEELSE
			PRINTFORMW 在长凳上一起享用了。
		ENDSELECT
	ENDIF
	CALL KOJO_DATE, "便当"
	FLAG:情緒 += 30
CASE "教会午餐"
	CALL KOJO_DATE, "教会午餐"
	FLAG:情緒 += 30
CASE "特别忏悔室"
	CALL KOJO_DATE, "特别忏悔室"
	FLAG:情緒 += 30
CASE "牵手"
	SIF TFLAG:牵手 || TFLAG:约会中牵手可能 == 0
		RETURN 0
	CALL BLANK_DRAWLINE
	PRINTFORMW 顺便一提。
	IF COND("合意：牵手")
		PRINTFORMW 突然意识到今天还没握过%CALLNAME:TARGET%的手。
	ELSEIF TALENT:相思相愛
		PRINTFORMW 突然意识到明明是一对恋人、却没有握住%CALLNAME:TARGET%的手。
	ELSE
		PRINTFORMW 突然意识到明明是在约会中、却没有握住%CALLNAME:TARGET%的手。
	ENDIF
	PRINTFORMW ……怎么办？
	IF COND("合意：牵手") || TALENT:恋慕
		CALL PRINT_SELECT, "牵手/还是算了"
	ELSE
		CALL PRINT_SELECT, "牵手/可能会被讨厌、还是算了"
	ENDIF

	;手をつないだ
	IF RESULT == 0
		TFLAG:牵手 = 1
		FLAG:情緒 += 30

		CALL KOJO_DATE, "手をつなぐ"
	ELSE
		CALL KOJO_DATE, "手をつながない"
	ENDIF
CASE "购物"
	SELECTCASE SAVESTR:约会
	CASE "去买泳衣"
		CALL EVENT_SHOP_SWIM
	CASE "尋訪美食", "逛街", "散歩"
		IF COND("今年の泳装選択済") == 0 && COND("池面") == 0 && COND("春") + COND("夏")
			CALL BLANK_DRAWLINE
			PRINTFORMW ……说起来、今年的泳装还没有买的样子。要怎么办呢？
			CALL PRINT_SELECT, "马上去买/今天就算了"
			IF RESULT == 0
				CALL EVENT_SHOP_SWIM
				SAVESTR:约会 = 去买泳衣
				TFLAG:约会中H可能 = 1
			ELSE
				PRINTFORMW 下次再去选泳装吧。
			ENDIF
		ENDIF
	ENDSELECT
CASE "约会H"
	;情緒が高まって来たらエロい事もできるよ
	EVENT_TABLE =
	IF TFLAG:约会中H可能 && (COND("発情期") || MOOD_HEART() >= 2 || SAVESTR:约会 == "去图书馆")
		SELECTCASE SAVESTR:约会
		CASE "去买泳衣"
			;SIF PENIS(MASTER) &&  BASE:MASTER:精力 >= 30 && BASE:MASTER:魂 >= 100 && EXP:性処理経験 && COND("女性主導生性交") && COND("性処理可能") && CANTRAIN("正常位")
			;	EVENT_TABLE += "借用一下前穴/"
			SIF PENIS(MASTER) &&  BASE:MASTER:精力 >= 30 && BASE:MASTER:魂 >= 100 && EXP:性処理経験 && EXP:精飲経験 > 5 && COND("性処理可能") && CANTRAIN("口交")
				EVENT_TABLE += "请求偷偷口交/"
			SIF EVENT_TABLE != ""
				EVENT_TABLE += "忍耐/"
		;2019/10 hokaguru 追記
		CASE "去电影院"
			SIF PENIS(MASTER) &&  BASE:MASTER:精力 >= 30 && BASE:MASTER:魂 >= 100 && EXP:性処理経験 && EXP:手淫経験 > 5 && COND("性処理可能") && CANTRAIN("手淫")
				EVENT_TABLE += "请求偷偷手交/"
			SIF EVENT_TABLE != ""
				EVENT_TABLE += "专心看电影/"
		;2022/12/16 hokaguru 追記
		CASE "在网咖度过"
			SIF PENIS(MASTER) &&  BASE:MASTER:精力 >= 30 && BASE:MASTER:魂 >= 100 && EXP:性処理経験 && EXP:手淫経験 > 5 && COND("性処理可能") && CANTRAIN("手淫")
				EVENT_TABLE += "请求偷偷手交/"
			SIF PENIS(MASTER) &&  BASE:MASTER:精力 >= 30 && BASE:MASTER:魂 >= 100 && EXP:性処理経験 && EXP:精飲経験 > 5 && COND("性処理可能") && CANTRAIN("口交")
				EVENT_TABLE += "请求偷偷口交/"
			SIF EVENT_TABLE != ""
				EVENT_TABLE += "忍耐/"
		;2020/02/24　玛丽酱の中の人　追記　雑に分岐を増やす
		CASE "去图书馆"
			SIF PENIS(MASTER) &&  BASE:MASTER:精力 >= 30 && BASE:MASTER:魂 >= 100 && EXP:性処理経験 && EXP:精飲経験 > 5 && COND("性処理可能") && CANTRAIN("口交")
				EVENT_TABLE += "请求偷偷口交/"
			;内容を考えるのが面倒なので一旦封印
;			SIF PENIS(MASTER) &&  BASE:MASTER:精力 >= 30 && BASE:MASTER:魂 >= 100 && EXP:性処理経験 && EXP:手淫経験 > 5 && COND("性処理可能") && CANTRAIN("手淫")
;				EVENT_TABLE += "请求偷偷手交/"
			SIF EVENT_TABLE != ""
				EVENT_TABLE += "认真学习/"
		;2021/07/27　玛丽酱の中の人が書いた
		CASE "温泉旅行"
			SIF PENIS(MASTER) &&  BASE:MASTER:精力 >= 30 && BASE:MASTER:魂 >= 100 && EXP:性処理経験 && EXP:精飲経験 > 5 && COND("性処理可能") && CANTRAIN("口交")
				EVENT_TABLE += "口交/"
			SIF PENIS(MASTER) &&  BASE:MASTER:精力 >= 30 && BASE:MASTER:魂 >= 100 && EXP:性処理経験 && EXP:手淫経験 > 5 && COND("性処理可能") && CANTRAIN("手淫")
				EVENT_TABLE += "手交/"
			SIF EVENT_TABLE != ""
				;ノータッチノー萝莉ータ
				;EVENT_TABLE += "紳士協定に基づき什麼也不做/"
				EVENT_TABLE += "什麼也不做/"
		CASE "自驾游约会"
			;温泉旅行のものを就这様パクりました
			SIF PENIS(MASTER) &&  BASE:MASTER:精力 >= 30 && BASE:MASTER:魂 >= 100 && EXP:性処理経験 && EXP:精飲経験 > 5 && COND("性処理可能") && CANTRAIN("口交")
				EVENT_TABLE += "口交/"
			SIF PENIS(MASTER) &&  BASE:MASTER:精力 >= 30 && BASE:MASTER:魂 >= 100 && EXP:性処理経験 && EXP:手淫経験 > 5 && COND("性処理可能") && CANTRAIN("手淫")
				EVENT_TABLE += "手交/"
			SIF EVENT_TABLE != ""
				EVENT_TABLE += "一旦休憩する/"
		CASE @"%CALLNAME:TARGET%的家"
			IF PENIS(MASTER) && BASE:MASTER:精力 >= 30 && BASE:MASTER:魂 >= 100 && EXP:精飲経験 > 5 && COND("性処理可能")
				SIF CANTRAIN("口交")
					EVENT_TABLE += "口交/"
				SIF CANTRAIN("手淫")
					EVENT_TABLE += "手交/"
			ENDIF
			SIF EVENT_TABLE != ""
				EVENT_TABLE += "什麼也不做/"
		CASEELSE
			SIF PENIS(MASTER) &&  BASE:MASTER:精力 >= 30 && BASE:MASTER:魂 >= 100 && EXP:性処理経験 && EXP:精飲経験 > 5 && COND("性処理可能") && CANTRAIN("口交")
				EVENT_TABLE += "到暗处去……/"
			SIF COND("合意：接吻")
				EVENT_TABLE += "接吻/"
			SIF CHECK_SKIRT("迷你") || EQUIP:裙 == 0 || TFLAG:泳装约会
				EVENT_TABLE += "摸屁股/"
			SIF EVENT_TABLE != ""
				EVENT_TABLE += "下次去哪里呢？询问/"
		ENDSELECT
	ENDIF

	SIF EVENT_TABLE == ""
		RETURN 0

	CALL BLANK_DRAWLINE
	SELECTCASE SAVESTR:约会
	CASE "去水族馆"
		PRINTFORMW ……\@ TFLAG:牵手 ? 手牵着手 # \@在幽暗的水族館里来回参观时、随着话语的减少、彼此间的距离缩短了。
	CASE "去游乐园"
		PRINTFORMW ……在热闹的主题公园\@ TFLAG:牵手 ? 手牵手 # \@来回参观时、随着话语的减少、彼此间的距离缩短了。
	CASE "去游泳"
		PRINTFORMW ……\@ TFLAG:牵手 ? 手牵着手 # \@穿着泳装開心地游玩时、随着话语的减少、彼此间的距离缩短了。
	CASE "去买泳衣"
		PRINTFORMW ……在狭窄的房间里、极近距离的泳装挑选结束后、%CALLNAME:MASTER%发现自己的股間变大了。
	CASE "去电影院"
		PRINTFORMW ……在幽暗的影厅内\@ TFLAG:牵手 ? 手牵着手 # \@观赏电影时、注意到%CALLNAME:TARGET%热切地看过来了。
	;2022/12/16 hokaguru 追記
	CASE "在网咖度过"
		PRINTFORMW \@ TFLAG:牵手 ? 緩くお互相用指を絡ませながら # \@%TEXTR("読んでいた漫画/見ていた页/パソコンの画面")%から目を離すと、%CALLNAME:TARGET%と目が合った。
		PRINTFORMW %CALLNAME:TARGET%は%TEXTR("すぐに/有点害臊的/頬を染めて/弾かれたように")%目を逸らしたが、先程よりも\@ TFLAG:牵手 ? 絡ませた指が強く握られ # \@やけに距離が縮まっている気がする…
	CASE "去图书馆"
		PRINTFORMW 注意力耗尽的%CALLNAME:MASTER%从书本中抬起頭来、和显然没有在看书的%CALLNAME:TARGET%对上了眼睛。
		PRINTFORMW %CALLNAME:TARGET%害羞地移開了眼睛、又再次用热腾腾的目光看向%CALLNAME:MASTER%……
	CASE "自驾游约会"
		PRINTFORM 在欣赏了一会周围的美景之后%CALLNAME:TARGET%们为了买饮料将车停在了
		SELECTCASE SAVESTR:現在位置
		CASE "街中"
			PRINTFORM 附近的便利店
		CASEELSE
			PRINTDATA
				DATA 休息区
				DATA 附近的车站边上
			ENDDATA
		ENDSELECT
		PRINTFORMW 边上。
		PRINTFORMW 一瞬间和%CALLNAME:TARGET%对上了视线、发现了对方正用热情又暧昧的目光看向%CALLNAME:MASTER%……
	;2021/07/27　玛丽酱の中の人が書いた
	CASE "温泉旅行"
		PRINTFORMW ……到达了%TEXTR("温泉旅館/预约的旅馆")%的%TEXTR(@"%CALLNAME:MASTER%和%CALLNAME:TARGET%/二人")%%TEXTR("＿/迅速/马上")%换上了浴衣。
		PRINTFORM 穿着浴衣的%CALLNAME:TARGET%
		;淫乱を分岐に入れると相思相愛の場合、ほぼ引っ掛かるので除外
		IF TALENT:喜歡引人注目 || TALENT:開朗 || TALENT:不知羞恥 || TALENT:粗野 || TALENT:好色 || TALENT:小悪魔
			PRINTFORMW 恶作剧似的将手放在%CALLNAME:MASTER%股间的浴衣上%TEXTR("開始按摩/開始抚摸")%。
		ELSE
			PRINTFORMW %TEXTR("或许是因为两人旅行而变得大胆了一些/或许是逐渐习惯了这种非日常的氛围")%、坐到了%CALLNAME:MASTER%的身边%TEXTR("做出了娇媚的姿态/用热切的目光看着这边")%。
		ENDIF
	CASEELSE
		PRINTFORMW ……\@ TFLAG:牵手 ? 手牵着手 # \@一起行动的同时、随着话语的减少、彼此间的距离缩短了。
	ENDSELECT

	VARSET LOCALS
	SPLIT EVENT_TABLE, "/", LOCALS
	CALL PRINT_SELECT, EVENT_TABLE
	CHOICES = %LOCALS:RESULT%

	SELECTCASE CHOICES
	CASE "认真学习"
		PRINTFORMW %CALLNAME:MASTER%们在稍微休憩了一会之后、又重新開始努力学习了。
		CALL SETFLAG_DAILY, "勉強"
	CASE "一旦休憩する"
		PRINTFORM %CALLNAME:MASTER%達は車を
		IF SAVESTR:現在位置 == "海" || SAVESTR:現在位置 == "山"
			PRINTFORM 停在了最近的停车场边、
		ELSE
			PRINTFORM 停在了附近的便利店边、
		ENDIF
		PRINTFORMW 在座椅之间開始了愉快的休息……。
	;2021/07/27　玛丽酱の中の人が書いた　口交を追加
	CASE "到暗处去……", "请求偷偷口交", "口交"
		CALL KOJO_DATE, CHOICES
		CLEARLINE 1

		CALL SIMULATE_TRAIN_FIRST
		MEMO_EJECT = EXP:MASTER:射精経験
		FOR LCOUNT, 0, 10
			CALL SIMULATE_TRAIN_COM, 91, 1
			;すでに射精済みならここでおしまい
			SIF EXP:MASTER:射精経験 > MEMO_EJECT
				BREAK
		NEXT
		CALL SIMULATE_TRAIN_CALC
		CALL SIMULATE_TRAIN_SHOW
		WAIT

		PRINTFORM %CALLNAME:TARGET%
		SELECTCASE SAVESTR:约会
		CASE "去买泳衣"
			PRINTFORM 为了尽量不弄脏付款前的泳装的样子、
		CASE "去游泳"
			PRINTFORM 因为是泳装所以有些污垢也没问题的样子、
		;2021/07/27　玛丽酱の中の人が書いた
		CASE "温泉旅行"
			IF TALENT:巨乳
				PRINTFORM 一边敞開胸部的浴衣展示%TEXTR("＿/胸部/欧派/乳房/乳沟")%、一边
			ELSE
				PRINTFORM 为了不弄脏%TEXTR("＿/刚刚换上的")%浴衣、
			ENDIF
		CASE "自驾游约会"
			PRINTFORM 因为不想把车弄脏
		CASEELSE
			IF COND("约会衣装に着替えた")
				PRINTFORM 专为约会而穿的
			ENDIF
			PRINTFORM 衣服不想弄脏的样子、
		ENDSELECT
		PRINTFORMW 用嘴接住了%CALLNAME:MASTER%的射精。

		IF COND("欲求不満") || TALENT:淫乱 || TALENT:好色 || COND("自発口交可能") || ABL:精液中毒 >= 3
			PRINTFORM 咕噜咕噜地蠕动喉咙%TEXTR("看起来很美味地/毫不排斥地/像理所当然一样")%饮下了精液的%CALLNAME:TARGET%、
			;手帕で口元を拭きながら
			PRINTFORM %TEXTR("一脸轻松地/用比行为前更有魅力的笑容/散发着刚出的汗的香味")%

			FLAG:情緒 += 30
		ELSE
			PRINTFORM %TEXTR("嗯哼……为/因为食物的/和以前一样因为")%味道和粘性而苦恼、很困难地喝下了口中的精液的%CALLNAME:TARGET%
			PRINTFORM %TEXTR("、用有点疲惫的样子/咳了一声调整好说话的声调、/从包里取出饮料洗刷喉咙、")%

			FLAG:情緒 -= 30
		ENDIF
		SELECTCASE SAVESTR:约会
		CASE "去买泳衣"
			PRINTFORMW 和%CALLNAME:MASTER%说要快点在店员发现之前付账离開……
		CASE "去游泳"
			PRINTFORMW 和%CALLNAME:MASTER%%TEXTR("聊起了関于下一个目的地的计划/问要不要再继续游泳/说这么热闹应该没有人会注意到吧/说还好这里是个不用担心脏污的好地方")%……
		;2022/12/16 hokaguru 追記
		CASE "在网咖度过"
			PRINTFORM %TEXTR("周囲を気にしながらも/口元を拭って")%
			IF COND("欲求不満") || TALENT:淫乱 || TALENT:好色
				PRINTFORM 続きはここを出てから
			ELSE
				PRINTFORM まだ退出には時間があるから
			ENDIF
			PRINTFORMW 、と漫画本を手に取った…
		CASE "去图书馆"
			PRINTFORMW %TEXTR("小心翼翼不发出声响地结束了……/摇着頭小声地问是不是可以结束了……")%
		CASE "自驾游约会"
			PRINTFORMW 再次握上了方向盘……
		CASE @"%CALLNAME:TARGET%的家"
			IF TALENT:内気 || TALENT:害羞 || CSTR:建议 == "理性主義"
				PRINTFORM 因为害羞红透了双颊
				SIF TALENT:淫乱
					PRINTFORM 、不时地用期待又诱惑的目光
				PRINTFORMW 看向%CALLNAME:MASTER%……
			ELSEIF TALENT:好色 || TALENT:淫乱 || TALENT:小悪魔
				PRINTFORM 用湿润的双眼盯住%CALLNAME:MASTER%的同时压了上来……
			ELSE
				PRINTFORMW 一边红着脸颊、一边又仿佛想要做什么事情一样地和%CALLNAME:MASTER%搭上了话……
			ENDIF
		;2021/07/27　玛丽酱の中の人が書いた
		CASE "温泉旅行"
			PRINTFORMW 小声地询问着%TEXTR("因为刚刚弄脏了身体要不要一起去泡温泉/想要清洁身体可不可以一起去泡温泉")%……
		CASEELSE
			PRINTFORMW 和%CALLNAME:MASTER%聊起了関于下一个目的地的计划……
		ENDSELECT

		TFLAG:口交 = 1

	;2019/10 hokaguru 追記　CALL SIMULATE_TRAIN_COMの指定指令を手淫に変えて↑を丸パク　一部地の文もパクパク美味しい
	CASE "请求偷偷手交"
		CALL KOJO_DATE, CHOICES
		CLEARLINE 1

		CALL SIMULATE_TRAIN_FIRST
		MEMO_EJECT = EXP:MASTER:射精経験
		FOR LCOUNT, 0, 10
			CALL SIMULATE_TRAIN_COM, 90, 1
			;すでに射精済みならここでおしまい
			SIF EXP:MASTER:射精経験 > MEMO_EJECT
				BREAK
		NEXT
		CALL SIMULATE_TRAIN_CALC
		CALL SIMULATE_TRAIN_SHOW
		WAIT

		IF COND("欲求不満") || TALENT:淫乱 || TALENT:好色 || ABL:精液中毒 >= 3
			PRINTFORMW %CALLNAME:TARGET%の手の中にある%TEXTR("阴茎/ちんぽ/モノ/剛直/欲望")%が%TEXTR("痙攣する/一際震える/先端を膨れさせる")%と、%TEXTR("粘りのある/クセのある气味の")%液体が吐き出された。
			PRINTFORM %TEXTR("危なげない手つきでそれを空いた手で受け止めた/吐き出させるままに扱き続けた/求めていたものが吹き出す瞬間を迎えた")%%CALLNAME:TARGET%は、
			PRINTFORM 手に付着した%TEXTR("精液/精液/スペルマ/ソレ/白濁")%を%TEXTR("舌先で舐め上げる/唇で拭うようにする/見て喉を鳴らす")%と、
			;手交なので口交に比べて低情緒補正で良いかな、と愚考。いや、どっちも公共の場でやるにはやべえけどね？
			FLAG:情緒 += 15
		ELSE
			PRINTFORM 場所を選ばず吐き出された%TEXTR("濁汁/白濁/精液/精液")%に一瞬%TEXTR("呆けた/顔をしかめた/驚いた様子の")%%CALLNAME:TARGET%は
			PRINTFORM %TEXTR("以一副微妙的表情/好像很没办法地/慌忙地")%用手帕擦干净了、
			;同上 口交よりましだ、なら情緒の下がり幅は狭窄はずだ（などと供述
			FLAG:情緒 -= 15
		ENDIF
		SELECTCASE SAVESTR:约会
		CASE "去电影院"
			IF COND("欲求不満") || TALENT:淫乱 || TALENT:好色
				PRINTFORMW 在耳边小声地说%TEXTR("「在哪里继续？」/「这就结束了吗？」/「过会继续」")%……
			;精液中毒持ちかつ喉を鳴らすとこんなこと言いながら拭ってません、どうしたのでしょうか。安らかに息を引き取る。
			ELSEIF TALENT:害羞
				PRINTFORMW 一边说着%TEXTR("「希望你能把注意力集中在电影上」/「在这种地方胡想些什么啊」")%、一边红着脸无可奈何地叹了口气……
			ELSE
				PRINTFORMW 在%CALLNAME:MASTER%耳旁悄悄地说「等到放映结束吧」……
			ENDIF
		;2023/02/20 道端の菠萝包 追記
		CASE "自驾游约会"
			IF COND("欲求不満") || TALENT:淫乱 || TALENT:好色
				PRINTFORMW %TEXTR("说着请期待接下来的事情/舔舐着指尖残留的精汁/红着脸微笑着")%吐出淫乱的喘息…
			ELSEIF TALENT:害羞
				PRINTFORM %TEXTR(@"似乎是很在意有没有被人发现一样/想要快点结束地催促起%CALLNAME:MASTER%")%
				IF ABL:精液中毒 >= 3
					PRINTFORMW 一边舔干净指尖的精液…
				ELSE
					PRINTFORMW 一边仔细地擦干净了射的到处都是的精液…
				ENDIF
			ELSE
				PRINTFORMW 在快速地结束事情之后、重新地開动了汽车…
			ENDIF
		;2022/12/16 hokaguru 追記
		CASE "在网咖度过"
			IF COND("欲求不満") || TALENT:淫乱 || TALENT:好色
				PRINTFORMW %TEXTR("淫蕩な露出笑容/情欲に浮かべた汗を拭いもせず/小さく舌を出して")%挑発的に終わりを囁いた…
			ELSEIF TALENT:害羞
				PRINTFORMW %TEXTR("周りに音が漏れていないか/臭気で気付かれていないか/周囲の様子が")%気になって仕方無いようだ…
			ELSE
				PRINTFORMW %TEXTR("小小的笑み浮かべ頷いて/しーっと人差し指を立てて/手早く後始末を終えて")%何事も無かったかのように傍らの漫画本に手を付けた…
			ENDIF
		;2021/07/27　玛丽酱の中の人が書いた
		CASE "温泉旅行"
			IF COND("欲求不満") || TALENT:淫乱 || TALENT:好色
				PRINTFORMW 在耳边嗫嚅着%TEXTR("接下来想在温泉里面/想要在温泉里面继续做/想去温泉里面继续")%……
			;精液中毒持ちかつ喉を鳴らすとこんなこと言いながら拭ってません、どうしたのでしょうか。安らかに息を引き取る。
			ELSEIF TALENT:害羞
				PRINTFORMW %TEXTR("明明才刚刚到达旅馆/明明还有很长的时间")%、这样说着红着脸的%CALLNAME:TARGET%带着稍微有点惊讶的表情叹了口气邀请一起去温泉里面做……
			ELSE
				PRINTFORMW 在%CALLNAME:MASTER%的耳边说着、%TEXTR("一起去泡一会温泉转换一下心情吧/一起去温泉里面洗洗身体吧")%……
			ENDIF

		CASEELSE
			IF COND("欲求不満") || TALENT:淫乱 || TALENT:好色
				PRINTFORMW %TEXTR("一副满意的表情/一边享受手上的残香一边")%聊起了関于下一个目的地的计划……
			ELSEIF TALENT:害羞
				PRINTFORMW %TEXTR("无法住控制害羞的表情/眼睛不知道该看向哪里")%、沉默地抓住%CALLNAME:MASTER%的手就离開了那个地方……
			ELSE
				PRINTFORMW 和%CALLNAME:MASTER%聊起了関于下一个目的地的计划……
			ENDIF
		ENDSELECT

		TFLAG:手淫 = 1

	CASE "接吻"
		PRINTFORMW %CALLNAME:MASTER%%TEXTR(@"自然地接近%CALLNAME:TARGET%的脸、接吻了/不意之间亲吻了沉默中的%CALLNAME:TARGET%/吻了无防备的%CALLNAME:TARGET%")%。
		;汎用地の文が無いのでMEMO_LINECOUNTで台詞の有無を見る
		MEMO_LINECOUNT = LINECOUNT
		CALL KOJO_DATE, CHOICES
		SIF LINECOUNT != MEMO_LINECOUNT
			CLEARLINE 1
		FLAG:情緒 += 30

		TFLAG:接吻 = 1
	CASE "摸屁股"
		CALL PRINT_CUTIN, "見たまま背面", TARGET
		PRINTFORM %CALLNAME:MASTER%把%CALLNAME:TARGET%的
		IF TFLAG:泳装约会
			IF CHECK_SKIRT("裙") || CHECK_SKIRT("荷葉邊")
				PRINTFORM %SKIRT(TARGET, "略")%の下に手を忍ばせ、%TEXTS("お尻：感触")%を掌で味わいながら
			ELSE
				PRINTFORM %TEXTS("お尻：感触")%を
				IF EQUIP:外套
					PRINTFORM 隔着%OUTER(TARGET, "略")%
				ELSEIF CHECK_PANTIES("Ｔ字褲")
					PRINTFORM 直接
				ELSE
					PRINTFORM 隔着泳装
				ENDIF
				PRINTFORM 用手掌品味着的同时、
			ENDIF
			IF CHECK_PANTIES("Ｔ字褲") && EQUIP:外套 == 0
				PRINTFORMW 指先%TEXTR("用被勒入的纽带在肛門里悪戯着/抚摸着紧贴纤细布料的股間")%。
			ELSEIF TALENT:男性
				PRINTFORMW 用手指%TEXTR("滑入了股間的缝隙/咕哩咕哩地玩弄着股間/从尻穴到会陰来回按摩着")%。
			ELSEIF CHECK_SKIRT("裙") || CHECK_SKIRT("荷葉邊")
				PRINTFORMW 用手指%TEXTR("隔着布料享受着柔软阴部的触感/伸到紧贴股間的布料下抚摸着柔软的阴部")%。
			ELSE
				PRINTFORMW 用手指%TEXTR("滑入了股間的缝隙/咕哩咕哩地玩弄着股間/从阴部到尻穴来回按摩着")%。
			ENDIF
		ELSE
			;裙ならば迷你のみ
			IF CHECK_SKIRT("迷你")
				PRINTFORM 迷你裙的下面用手抚摸着、一边用手掌品味%TEXTS("お尻：感触")%
			ELSE
				PRINTFORM %TEXTS("お尻：感触")%を
				IF EQUIP:外套
					PRINTFORM 隔着%OUTER(TARGET, "略")%
				ELSEIF EQUIP:贴身制服
					PRINTFORM 隔着%BODYS(TARGET, "略")%
				ELSEIF EQUIP:胖次
					PRINTFORM 隔着%PANTIES(TARGET, "胖次")%
				ENDIF
				PRINTFORM 用手掌品味着的同时、
			ENDIF
			IF COND("胖次形状確認")
				IF EQUIP:胖次 == 0 || CHECK_PANTIES("Ｖ露出")
					IF TALENT:男性
						PRINTFORMW 用手指%TEXTR("咕哩咕哩地玩弄着裸露的股間/从尻穴到会陰来回按摩着")%。
					ELSE
						PRINTFORMW 用手指%TEXTR("咕哩咕哩地玩弄着裸露的股間/从阴部到尻穴里的粘膜来回抚摸着")%。
					ENDIF
				ELSEIF CHECK_PANTIES("Ｔ字褲") && CHECK_SKIRT("迷你")
					PRINTFORMW 用手指%TEXTR("越过Ｔ字褲的纽带在肛門里悪戯着/抚摸着紧贴纤细布料的股間")%。
				ELSEIF TALENT:男性
					PRINTFORMW 用手指%TEXTR("滑入了股間的缝隙/咕哩咕哩地玩弄着股間/从尻穴到会陰来回按摩着")%。
				ELSEIF CHECK_SKIRT("迷你")
					PRINTFORMW 用手指%TEXTR("隔着胖次享受着柔软阴部的触感/伸到胖次下抚摸着柔软的阴部")%。
				ELSE
					PRINTFORMW 用手指%TEXTR("滑入了股間的缝隙/咕哩咕哩地玩弄着股間/从阴部到尻穴来回按摩着")%。
				ENDIF
			ELSE
				IF TALENT:男性
					PRINTFORMW 用手指%TEXTR("滑入了股間的缝隙/咕哩咕哩地玩弄着股間")%。
				ELSE
					PRINTFORMW 用手指%TEXTR("滑入了股間的缝隙/咕哩咕哩地玩弄着股間/从阴部到尻穴来回按摩着")%。
				ENDIF
				IF EQUIP:裙 && CHECK_CLO("胖次視認可能")
					PRINTFORM 从手感上来说、今天的%CALLNAME:TARGET%……
					IF EQUIP:胖次 == 0
						PRINTFORMW 竟然、没有穿胖次！
					ELSEIF CHECK_PANTIES("前貼")
						PRINTFORMW 穿的是贴在股間上的C字裤。
					ELSEIF CHECK_PANTIES("兜襠布")
						PRINTFORMW 好像系着兜襠布。
					ELSE
						PRINTFORMW 穿的应该是%DETAIL_PANTIES(TARGET, "感触")%。
					ENDIF
				ENDIF
			ENDIF
		ENDIF
		CALL KOJO_DATE, CHOICES
		CLEARLINE 1
		FLAG:情緒 += 15

		TFLAG:摸屁股 = 1
	CASE "忍耐"
		PRINTFORMW 在这里不适合做这种事、所以忍了下来。
	CASEELSE
		CALL KOJO_DATE, CHOICES
		IF TFLAG:地の文スキップ == 0
			SELECTCASE CHOICES
			CASE "专心看电影"
				PRINTFORMW %CALLNAME:TARGET%看了看专心看电影的%CALLNAME:MASTER%的侧脸、将注意力重新集中到了电影上……
			;2021/07/27　玛丽酱の中の人が書いた
			CASE "什麼也不做"
				PRINTFORMW %CALLNAME:TARGET%坐到%CALLNAME:MASTER%的身边、%TEXTR(@"读着小册子提醒%CALLNAME:MASTER%这里有/邀请%CALLNAME:MASTER%一起去")%特别设置在旅馆房间中的个人温泉……
			CASEELSE
				PRINTFORMW %CALLNAME:TARGET%出乎意料的很享受和%CALLNAME:MASTER%的约会、什么都没说地把身体靠了过来……
			ENDSELECT
		ENDIF
		FLAG:情緒 += 15
	ENDSELECT
	CALL BLANKLINE
CASE "晒痕"
	SIF TFLAG:泳装约会 == 0
		RETURN 0
	;どっちかが晒痕したか判定
	SIF COND("晒痕可能") == 0 && COND("晒痕可能", MASTER) == 0
		RETURN 0

	CALL BLANK_DRAWLINE
	PRINTFORMW 泳装约会的%CALLNAME:MASTER%\@ COND("晒痕可能") ? # 一行人 \@得到了晒痕。

	IF COND("晒痕可能")
		CFLAG:晒痕強度 = CALCF("晒痕強度")
		CSTR:晒痕 = %SWIM(CSTR:今年の泳装, "晒痕")%
	ENDIF

	IF COND("晒痕可能", MASTER)
		CFLAG:MASTER:晒痕強度 = CALCF("晒痕強度")
		IF SAVESTR:泳装 != ""
			CSTR:晒痕 = %SWIM(SAVESTR:泳装, "晒痕")%
		ELSEIF TALENT:MASTER:男性
			CSTR:MASTER:晒痕 = 海胖次＆
		ELSE
			CSTR:MASTER:晒痕 = 比基尼＆比基尼胸罩
		ENDIF
	ENDIF
CASE "文化祭"
	CALL BLANK_DRAWLINE

	SELECTCASE TFLAG:事件回数
	CASE 0
		PRINTFORMW 那样的话……值得記念的首次要去哪里呢？
	CASE 1
		PRINTFORMW ……接下来要去看哪项呢？
	CASE 2
		PRINTFORMW ……道路上的人数已经渐行渐少了。最後要去哪里呢？
	ENDSELECT

	STR:選択肢 = %SHOP_BUNKASAI()%
	VARSET LOCALS
	SPLIT STR:選択肢, "/", LOCALS
	CALL PRINT_SELECT, STR:選択肢
	CHOICES = %LOCALS:RESULT%

	PRINTFORM %CALLNAME:MASTER%和%CALLNAME:TARGET%去前往参观了%CHOICES%
	;人いるか格子
	ACTOR = FIND_COND(CHOICES)
	IF ACTOR
		PRINTFORMW ……発見了%CALLNAME:ACTOR%！

		;partnerを重置＆設定
		CALL RESET_SET_PARTNER, ACTOR
		;本来のTARGETをpartner、ACTORをTARGETとしてイベント進行させる
		CALL SWAP_PARTNER_TARGET
		;ふさわしい服装に着替えさせる
		CALL SET_COORDINATE_BUNKASAI, CHOICES, TARGET
		CALL PRINT_MAIDFACE, TARGET
		CALL MESSAGE_BUNKASAI, CHOICES
		;出演済みフラグを立てる
		CALL SET_CEVENT, "学園祭：出演済"
		CALL SWAP_PARTNER_TARGET

		;partnerを再セット
		CALL SET_PARTNER
	ELSE
		PRINTFORMW ……会是怎样的呢？
	ENDIF

	;この場所を巡ったフラグを立てる
 	CALL SET_CEVENT, @"学園祭：%CHOICES%", MASTER
	TFLAG:事件回数 += 1
CASE "约会撮影"
	;いっぱいなら中止
	SIF NUM("写真") >= 300
		RETURN 0
	CALL BLANK_DRAWLINE
	PRINTFORMW 似乎是很享受约会、%CALLNAME:TARGET%的表情也逐渐丰富了起来…
	PRINTFORMW 想拍张照片来纪念这次的约会吗？
	CALL PRINT_SELECT, "拍/不拍"
	IF RESULT == 0
		CALL PHOTO_DATE
	ELSE
		PRINTFORMW 将%CALLNAME:TARGET%的笑顔牢牢地烙印在了心中。
	ENDIF
ENDSELECT

IF FLAG:情緒 > MEMO_MOOD
	CALL TEXT, "情緒変化＆爱心個数表示", FLAG:情緒 - MEMO_MOOD
	WAIT
ENDIF

;-------------------------------------------------
;泳装屋。ARGSに"自動選択"が入っていると随意選びます
;-------------------------------------------------
@EVENT_SHOP_SWIM, ARGS
#DIMS MEMO_PLACE

MEMO_PLACE = %SAVESTR:現在位置%
SAVESTR:現在位置 = 試着室
CALL SETFLAG, "衣装の記録", TARGET

;現在的服装を記録したうえでとりあえず比基尼に更衣
VARSET LOCALS
SIF OTHER(TARGET, "眼鏡") != ""
	LOCALS += @"%OTHER(TARGET, "眼鏡")%/"
SIF OTHER(TARGET, "髪飾") != ""
	LOCALS += @"%OTHER(TARGET, "髪飾")%/"
CALL SETFLAG, "衣装全脱ぎ", TARGET
CALL SET_OTHER, LOCALS, TARGET

IF COND("池面")
	CALL SET_PANTIES, "衝浪胖次", TARGET
	CALL SET_BRA, "", TARGET
	STR:LASTCHOICE = 胖次_衝浪胖次
ELSE
	CALL SET_PANTIES, "比基尼", TARGET
	CALL SET_BRA, @"比基尼胸罩/%CLOCOLOR(CSTR:胖次)%", TARGET
	STR:LASTCHOICE = 胖次_比基尼
ENDIF
CALL SET_SHOES, "凉鞋", TARGET

STR:選んだ泳装 =

SELECTCASE ARGS
CASE "自動選択"
	CALL SETFLAG, "今年の泳装の記録", TARGET
CASEELSE
	CALL BLANK_DRAWLINE
	SELECTCASE SAVESTR:约会
	CASE "去买泳衣"
		PRINTFORMW 差不多是今天最重要的事、%CALLNAME:TARGET%的泳装挑选開始了。
	CASEELSE
		PRINTFORMW %CALLNAME:MASTER%到达了附近的泳装店。
	ENDSELECT

	CALL BLANKLINE
	CALL KOJO_DATE, "泳装選び開始"
	IF CSTR:今年の泳装 != ""
		PRINTFORMW 現在%CALLNAME:TARGET%拥有的是%SWIM(CSTR:今年の泳装)%、打算怎么办呢？
	ELSE
		PRINTFORMW 两人在店内挑了许久后、有了以下几个候选方案……
	ENDIF
	CALL PRINT_STRW, "※_緑_[比基尼or贴身制服]_に加えて、_水色_[外衣～裙]_の中から一つを合わせることが出来ます※"
	;外見特徴を羅列
	PRINTFORM 顺便一提%CALLNAME:TARGET%是
	CALL PRINT_LIST, SWIM_APP(TARGET), "拡大表示"
	PRINTFORMW 。

	;泳装を選ぶ
	CALL SIMULATE_SWIM_BUY
	CALL BLANKLINE
	CALL KOJO_DATE, "泳装決定"
	CALL PRINT_STRW, @"%CALLNAME:MASTER%和%CALLNAME:TARGET%在试穿和商量之后、选中了_黄色_%CLO_ABOUT(TARGET)%_作为今年的泳衣。"
	FLAG:情緒 += 30
ENDSELECT

CSTR:今年の泳装 = %STR:選んだ泳装%
CFLAG:泳装選び = FLAG:年

;服を元に戻す
CALL SETFLAG, "衣装記録からの復元", TARGET
STR:衣装記録 =
SAVESTR:現在位置 = %MEMO_PLACE%


code:
;-------------------------------------------------
;朝に起こるイベント
;Ｈイベントの条件を満たしていなくても添い寝は確実に発生する
;基本的な流れとしては以下の順番で構成されている。
;1 朝Ｈ地の文(導入)						CALL BEFORETEXT_NEWDAY
;2 朝袭台詞							CALL KOJO_GREETING, "朝袭", ARG
;3 経験変動一覧の表示					CALL SIMULATE_SHOW_NEWDAY
;4 朝Ｈ地の文(事後)						CALL AFTERTEXT_NEWDAY
;5 颜绘(下の打招呼台詞直前に自動出力)	CALL EMOTION_PLUS, "甜言蜜语", ARG
;6 打招呼台詞
;
;～地の文とか色々消して自分の好きに書きたい！～
;2の朝袭台詞内で CALL SETFLAG, "地の文スキップ" とすると1～6を消したうえで表示しなく出来る
;この場合は対応した1,3,4,5に対応した右側の関数を自己設定しつつ自由にできる…が上級者向け
;（詩燈寧子、青空鈴などで記述例が見られます）
;-------------------------------------------------
@EVENT_NEWDAY, ARGS
#DIM LCOUNT
#DIM MEMO_LINECOUNT
#DIM MEMO_ABL
#DIM MEMO_EXP

;朝Ｈフラグの重置
CVARSET CSTR, GETNUM(CSTR, "朝袭")
;朝袭の衣装にお着替え
CALL SET_COORDINATE, "朝袭衣装"

;最初に施虐属性イベントの判定を行う(発生確率が低い)。内容は CSTR:朝袭 に入っている
CALL BIND_AT_MORNING
IF RESULT == 1 && ARGS == ""
	;朝拘束発生フラグON
	SETBIT FLAG:状況判定, 14
	BEGIN TURNEND
	RETURN 1
ENDIF

;朝イベントの決定。内容は CSTR:朝袭 に入っている
CALL PLAY_AT_MORNING
SIF ARGS != ""
	CSTR:朝袭 = %ARGS%

;基本回数は精力依存
TFLAG:PLAY回数 = LIMIT(BASE:精力/20 + LIMIT(RAND:(BASE:精力/10 + 1), 1, BASE:精力/15), 1, 3)
;イベントごとの初期設定
SELECTCASE CSTR:朝袭
CASE "朝逆正常位"
	TFLAG:PLAY回数 += MIN(3, ABL:欲望/3)
	SELECTCOM = TNUM("正常位奉仕")
CASE "朝舐陰"
	TFLAG:PLAY回数 += MIN(3, ABL:奉仕精神/3)
	SELECTCOM = TNUM("奉仕舐陰")
CASE "朝足交"
	TFLAG:PLAY回数 += MIN(3, ABL:施虐属性/3)
	SELECTCOM = TNUM("足交")
CASE "朝の新妻PLAY"
	TFLAG:PLAY回数 += MIN(5, (ABL:奉仕精神 + ABL:露出癖)/5)
	IF TALENT:男性
		SELECTCOM = TNUM("肛門後背位")
	ELSE
		SELECTCOM = TNUM("後背位")
	ENDIF
CASE "朝乳交"
	TFLAG:PLAY回数 += MIN(3, (ABL:奉仕精神 + ABL:精液中毒)/5)
	SELECTCOM = TNUM("乳交")
CASE "朝面部騎乗"
	TFLAG:PLAY回数 = LIMIT(TFLAG:PLAY回数/3 + ABL:施虐属性/3, 1, 1 + TALENT:相思相愛 + TALENT:淫核)
	SELECTCOM = TNUM("舐陰")
CASE "朝騎乗位"
	TFLAG:PLAY回数 += MIN(3, (ABL:Ｖ感覚 + ABL:精液中毒)/3)
	SELECTCOM = TNUM("騎乗位")
CASE "朝Ａ騎乗位"
	TFLAG:PLAY回数 += MIN(3, (ABL:Ａ感覚 + ABL:精液中毒)/3)
	SELECTCOM = TNUM("肛門騎乗位")
CASE "朝接吻"
	TFLAG:PLAY回数 = LIMIT(TFLAG:PLAY回数/2 + (ABL:信頼 + ABL:奉仕精神)/5, 1, 1 + TALENT:相思相愛 + TALENT:蕩唇)
	SELECTCOM = TNUM("接吻")
CASE "朝添い寝"
	TFLAG:PLAY回数 = 0
	SELECTCOM = -1
CASE "朝口交"
	TFLAG:PLAY回数 += MIN(3, (ABL:奉仕精神 + ABL:精液中毒)/3)
	SELECTCOM = TNUM("口交")
CASE "朝六九式"
	TFLAG:PLAY回数 += MIN(3, (ABL:奉仕精神 + ABL:精液中毒)/3)
	SELECTCOM = TNUM("六九式")
ENDSELECT

;複数PLAYの場合はここ。CSTR:朝袭はTARGETの方のみ先頭に"Ｗ"が付いている
SELECTCASE CSTR:朝袭
CASE "Ｗ朝口交"
	;partnerに設定
	CALL SET_PARTNER, FIND_COND("Ｗ朝口交可能", "相性○")
	;TARGET:1は経験表示の二人目なので設定しておく
	TARGET:1 = PARTNER()

	CSTR:PARTNER():朝袭 = 朝口交
	TFLAG:PLAY回数 += MIN(3, (ABL:奉仕精神 + ABL:精液中毒)/3) + MIN(3, (ABL:PARTNER():奉仕精神 + ABL:PARTNER():精液中毒)/3)
	SELECTCOM = TNUM("口交")
	;こっちもお着替え
	CALL SET_COORDINATE, "朝袭衣装", PARTNER()
ENDSELECT

MEMO_ABL = ABL:MASTER:技巧
;寝ているので技巧は0扱いにする
ABL:MASTER:技巧 = 0

;前準備開始
CALL SIMULATE_TRAIN_FIRST, 1 + (PARTNER() > 0)

IF TFLAG:PLAY回数
	IF PARTNER()
		MEMO_EXP = EXP:PLAYER:射精経験
		CALL SIMULATE_TRAIN_COM, SELECTCOM, MAX(1, TFLAG:PLAY回数*2/3)
		;你が射精しない時にはBOUNS游戏発生
		FOR LCOUNT, 0, 5
			SIF EXP:PLAYER:射精経験 > MEMO_EXP
				BREAK
			CALL SIMULATE_TRAIN_COM, SELECTCOM, 1
		NEXT

		CALL SWAP_PARTNER_TARGET
		MEMO_EXP = EXP:PLAYER:射精経験
		CALL SIMULATE_TRAIN_COM, SELECTCOM, MAX(1, TFLAG:PLAY回数/2)
		FOR LCOUNT, 0, 5
			SIF EXP:PLAYER:射精経験 > MEMO_EXP
				BREAK
			CALL SIMULATE_TRAIN_COM, SELECTCOM, 1
		NEXT
		CALL SWAP_PARTNER_TARGET
	ELSE
		CALL SIMULATE_TRAIN_COM, SELECTCOM, TFLAG:PLAY回数
	ENDIF
ENDIF

REDRAW 0
DRAWLINE
;地の文をスキップする場合の準備
TFLAG:地の文スキップ = 0
TFLAG:LINECOUNT_地の文 = LINECOUNT

;1 朝Ｈ地の文(導入)。ここはスキップに備えてWAITを一切使ってない。
CALL BEFORETEXT_NEWDAY

;2 朝袭台詞。ここでCALL SETFLAG, "地の文スキップ"すると3と4と5をすっ飛ばすので全部自己CALLするか書く必要がある
MEMO_LINECOUNT = LINECOUNT
CALL KOJO_GREETING, "朝袭", TARGET
;地の文スキップされていないなら默认の流れ
IF TFLAG:地の文スキップ == 0
	;もしもKOJO_GREETING, "朝袭"で台詞没有＆地の文スキップ没有ならば、BEFORETEXT_NEWDAYでWAITを使ってないので一回WAITを噛ませる
	SIF LINECOUNT == MEMO_LINECOUNT
		WAIT

	;3 経験変動一覧の表示
	CALL SIMULATE_SHOW_NEWDAY
	;4 朝Ｈ地の文(事後)
	CALL AFTERTEXT_NEWDAY
	;5&6 颜绘と打招呼台詞
	;複数PLAYの場合はここ
	IF PARTNER()
		;PARTNERはCSTR:朝袭 = 朝口交、TARGETはCSTR:朝袭 = Ｗ朝口交 となっているため喋る内容が違うのがミソ
		CALL SWAP_PARTNER_TARGET
		CALL KOJO_GREETING, "打招呼", TARGET
		CALL SWAP_PARTNER_TARGET
	ENDIF

	CALL KOJO_GREETING, "打招呼", TARGET
ENDIF

;技巧を元に戻す
ABL:MASTER:技巧 = MEMO_ABL

;欲求不満度の解消度の尺寸で分ける
SELECTCASE CSTR:朝袭
CASE "朝逆正常位", "朝の新妻PLAY", "朝騎乗位", "朝Ａ騎乗位"
	CALL CALC, "欲求不満解消", TARGET, MIN(TFLAG:PLAY回数*25, 75)
CASE "朝口交", "朝六九式", "朝面部騎乗"
	CALL CALC, "欲求不満解消", TARGET, MIN(TFLAG:PLAY回数*20, 50)
CASE "朝舐陰", "朝足交", "朝乳交", "朝接吻"
	CALL CALC, "欲求不満解消", TARGET, MIN(TFLAG:PLAY回数*10, 50)
CASE "Ｗ朝口交"
	CALL CALC, "欲求不満解消", TARGET, MIN(TFLAG:PLAY回数*10, 50)
	CALL CALC, "欲求不満解消", PARTNER(), MIN(TFLAG:PLAY回数*10, 50)
ENDSELECT

;-------------------------------------------------
;経験変動一覧の表示
;-------------------------------------------------
@SIMULATE_SHOW_NEWDAY
#DIM MEMO_LINECOUNT

MEMO_LINECOUNT = LINECOUNT

CALL SIMULATE_TRAIN_CALC
CALL SIMULATE_TRAIN_SHOW

SIF LINECOUNT > MEMO_LINECOUNT
	WAIT

;-------------------------------------------------
;朝Ｈ地の文(導入)
;-------------------------------------------------
@BEFORETEXT_NEWDAY
SELECTCASE CSTR:朝袭
CASE "朝逆正常位"
	PRINTFORML %CALLNAME:MASTER%因为感觉到下腹部\@ RAND:2 ? 襲来的衝撃 # 扩散開的快感 \@而醒了过来
	PRINTFORML 看起来、%CALLNAME:TARGET%正在享受着侵犯睡着的%CALLNAME:MASTER%的楽趣…
CASE "朝舐陰"
	PRINTFORML 早上、一醒过来就感觉股間好温暖…
	PRINTFORM 难道说、这么想着于是掀開了被子、结果和
	IF ABL:欲望 >= 5 && RAND:3 == 0
		PRINTFORM 非常兴奋的様子的
	ELSEIF RAND:3 == 0
		PRINTFORM 臉頰深深埋在股間的
	ELSEIF RAND:2
		PRINTFORM 把双腿大大地拉開
	ELSE
		PRINTFORM 把身体塞进双脚之間
	ENDIF
	PRINTFORML 、正舔着阴部的%CALLNAME:TARGET%对上了眼睛
CASE "朝足交"
	PRINTFORML 早上、因为%CALLNAME:TARGET%\@ PENIS(MASTER) ? 的足交 # 在用脚玩弄阴部 \@而醒了过来
CASE "朝の新妻PLAY"
	PRINTFORML 早上起来、从廚房飘来一股香味
	PRINTFORML 到了之后发现%CALLNAME:TARGET%在穿着裸体圍裙做早饭…
	PRINTFORML ―――衝動的从背后襲击
CASE "朝乳交"
	PRINTFORML 早上、醒来的同時从下腹部感觉到了違和感
	;貧乳の場合
	IF TALENT:貧乳 || TALENT:男性
		PRINTFORML 用不会被注意到的余光瞥了一眼、%CALLNAME:TARGET%在用胸部蹭着肉棒
		SIF EXP:乳交経験 < 50
			PRINTFORML 看起来是在做乳交的練習……
	;巨乳の場合
	ELSEIF TALENT:巨乳
		PRINTFORML 仔细一看%CALLNAME:TARGET%在用丰满的胸部夹着阴茎、不断摩擦
	;それ以外
	ELSE
		PRINTFORML 仔细一看%CALLNAME:TARGET%在用丰满的胸部夹着阴茎、并不停吮吸着前端流出的液体
	ENDIF
CASE "朝面部騎乗"
	PRINTFORML 早晨、%CALLNAME:MASTER%因为感觉呼吸困难而睁開了双眼
	IF TALENT:巨尻
		PRINTFORML 回过神来才发现整张脸被柔软的巨尻包裹着、滑滑的没有一点间隙地被尻肉压迫着…
	ELSEIF TALENT:小尻
		PRINTFORML 回过神来才发现整张脸紧致的小尻压着、滋滋地被尻肌摩擦着脸颊…
	ELSE
		PRINTFORML 回过神来才发现整张脸柔软的桃尻坐着、紧紧地与对方的尻肌贴在一起…
	ENDIF
	PRINTFORML %TEXTR("稍微有些/用没有任何防备的姿势")%%TEXTR("坏心眼地/被勾起了嗜虐心地")%、被%CALLNAME:TARGET%用面部騎乗%CALLNAME:MASTER%叫起来了

	IF CHECK_CLO("胖次露出")
		CALL SETFLAG_DAILY, "掀起裙子"
		CALL PRINT_CUTIN, "見たまま正面", TARGET
		CALL PRINT_STRL, @"冷静下来之后%CALLNAME:MASTER%顺便进行了检查……今日的%CALLNAME:TARGET%穿着的是%PANTIES(TARGET)%的样子"
	ENDIF
	SIF CHECK_CLO("Ｖ露出") && TEXTS("Ｖコンディション", TARGET) != ""
		CALL PRINT_STRL, @"(现在的%CALLNAME:TARGET%的Ｖ状况是_黄色_%TEXTS("Ｖコンディション", TARGET)%_)"

CASE "朝騎乗位"
	PRINTFORML 早上起来感觉腰上有什么东西压着
	PRINTFORM 睁開眼睛就发现%CALLNAME:TARGET%
	IF SIZE("阴茎", PLAYER) >= 2 && RANDIF(1, TALENT:巨乳)
		IF ABL:Ｖ拡張 >= SIZE("阴茎", PLAYER)
			PRINTFORM 跨在上面、把
			IF STRCOUNTS(CONDS("体型"), "苗条", "高挑")
				PRINTFORM 纤细的身体把
			ELSEIF TALENT:娇小
				PRINTFORM 小小的身体
			ENDIF
		ENDIF
		PRINTFORM %CALLNAME:MASTER%的
		SELECTCASE SIZE("阴茎", PLAYER)
		CASE IS >= 4
			PRINTFORM 規格外的巨根
		CASEELSE
			PRINTFORM 巨根
		ENDSELECT
		PRINTFORM \@ABL:Ｖ拡張 >= SIZE("阴茎", PLAYER) ? 用膣一直吞到最根部 # 压在胯下摩擦 \@
	ELSE
		PRINTFORM 跨在上面
		IF TALENT:巨乳
			IF RAND:3 == 0
				PRINTFORM 、%TOFULL(CONDS("罩杯") )%罩杯的%TALENT_NAME(GETNUM(TALENT, "巨乳"), TALENT:巨乳)%揺晃着
			ELSEIF TALENT:巨乳 >= 2
				IF RAND:2
					PRINTFORM 、豊満的胸部夸张地摇晃着
				ELSE
					PRINTFORM 、看上去很重的胸部悠悠地摇晃着
				ENDIF
			ELSEIF RAND:2
				PRINTFORM 、大大的胸部摇晃着
			ELSE
				PRINTFORM 、一边揉着自己的胸
			ENDIF
		ENDIF
	ENDIF
	PRINTFORML 发出了舒服的喘息……
	IF TARGET == ASSI
		IF RAND:2
			PRINTFORML 发现%CALLNAME:MASTER%醒过来之后、%CALLNAME:TARGET%緩緩動起了腰、一边夹杂着嬌声一边開始说明今天的日程…
		ELSE
			PRINTFORML 刚刚睡醒頭脑还处于混乱状态下的%CALLNAME:MASTER%、一边抓着%CALLNAME:TARGET%的腰向上突进、一边让其在嬌声说明今天的日程…
		ENDIF
	ENDIF
CASE "朝Ａ騎乗位"
	PRINTFORML 早上起来感觉腰上有什么东西压着
	PRINTFORM 睁開眼睛就发现%CALLNAME:TARGET%
	IF SIZE("阴茎", PLAYER) >= 2 && RANDIF(1, TALENT:巨乳)
		IF ABL:Ａ拡張 >= SIZE("阴茎", PLAYER)
			PRINTFORM 跨在上面、把
			IF STRCOUNTS(CONDS("体型"), "苗条", "高挑")
				PRINTFORM 纤细的身体把
			ELSEIF TALENT:娇小
				PRINTFORM 小小的身体
			ENDIF
		ENDIF
		PRINTFORM %CALLNAME:MASTER%的
		SELECTCASE SIZE("阴茎", PLAYER)
		CASE IS >= 4
			PRINTFORM 規格外的巨根
		CASEELSE
			PRINTFORM 巨根
		ENDSELECT
		PRINTFORM \@ABL:Ａ拡張 >= SIZE("阴茎", PLAYER) ? 用菊穴一直吞到最根部 # 压在胯下摩擦 \@
	ELSE
		PRINTFORM 跨在上面
		IF TALENT:巨乳
			IF RAND:3 == 0
				PRINTFORM 、%TOFULL(CONDS("罩杯") )%罩杯的%TALENT_NAME(GETNUM(TALENT, "巨乳"), TALENT:巨乳)%揺晃着
			ELSEIF TALENT:巨乳 >= 2
				IF RAND:2
					PRINTFORM 、豊満的胸部夸张地摇晃着
				ELSE
					PRINTFORM 、看上去很重的胸部悠悠地摇晃着
				ENDIF
			ELSEIF RAND:2
				PRINTFORM 、大大的胸部摇晃着
			ELSE
				PRINTFORM 、一边揉着自己的胸
			ENDIF
		ENDIF
	ENDIF
	PRINTFORML 发出了舒服的喘息……
	IF TARGET == ASSI
		IF RAND:2
			PRINTFORML 发现%CALLNAME:MASTER%醒过来之后、%CALLNAME:TARGET%緩緩動起了腰、一边夹杂着嬌声一边開始说明今天的日程…
		ELSE
			PRINTFORML 刚刚睡醒頭脑还处于混乱状态下的%CALLNAME:MASTER%、一边抓着%CALLNAME:TARGET%的腰向上突进、一边让其在嬌声说明今天的日程…
		ENDIF
	ENDIF
CASE "朝接吻"
	;奉仕レベルが高いと、先に起きている確率が高い
	IF ABL:奉仕精神 >= 5 && RANDF(2)
		PRINTFORML 早上、被%CALLNAME:TARGET%\@ TFLAG:PLAY回数 >= 3 ? 以濃厚的吻 # \@吻醒了
	ELSE
		IF (COND("夜袭相手") || AGE(TARGET) == 1) && RAND:2
			PRINTFORML 早上、%CALLNAME:MASTER%因为突然的揺晃和身上的重量而醒了过来
			PRINTFORML 看起来、是因为%CALLNAME:TARGET%跨上来还在晃晃悠悠地摇着%TEXTR("胸部/腰")%的缘故…
			PRINTFORML 注意到%CALLNAME:MASTER%醒过来的%CALLNAME:TARGET%把脸凑近、来了个早安吻
		ELSEIF TALENT:巨乳 && RAND:2
			PRINTFORML 早上、一醒过来就发现好像枕着什么非常舒服的东西……
			IF TALENT:不知羞恥 && RAND:3 == 0
				PRINT 全裸的
			ELSEIF RAND:2
				PRINT 只穿着内衣的
			ENDIF
			PRINTFORML %CALLNAME:TARGET%悄悄地跑到床上、把%CALLNAME:MASTER%的頭抱起来枕在了自己的胸上
			PRINTFORML 注意到%CALLNAME:MASTER%醒过来的%CALLNAME:TARGET%把脸凑近、来了个早安吻
		ELSE
			PRINTFORML 早上、一醒过来就感觉身体好重……
			PRINTFORML 不知道什么时候跑到床上的%CALLNAME:TARGET%、把%CALLNAME:MASTER%当做抱枕了的様子
			IF TALENT:相思相愛 && RAND:2
				PRINTFORML %CALLNAME:MASTER%因为无法抵抗心中涌起的愛意、对着%CALLNAME:TARGET%把嘴印了上去、
				PRINTFORML %CALLNAME:TARGET%好像很舒服似的抓住%CALLNAME:MASTER%的脸、深深地反亲了回来…
			ELSE
				PRINTFORML 注意到%CALLNAME:MASTER%醒过来的%CALLNAME:TARGET%把脸凑近、来了个早安吻
			ENDIF
		ENDIF
	ENDIF
CASE "朝添い寝"
	IF (COND("夜袭相手") || AGE(TARGET) == 1) && RAND:2
		PRINTFORML 早上、%CALLNAME:MASTER%因为突然的揺晃和身上的重量而醒了过来
		PRINTFORML 看起来、是因为%CALLNAME:TARGET%跨上来还在晃晃悠悠地摇着胸部的缘故…
	ELSEIF TALENT:巨乳 && RAND:2
		PRINTFORML 早上、一醒过来就发现好像枕着什么非常舒服的东西……
		IF TALENT:不知羞恥 && RAND:3 == 0
			PRINT 全裸的
		ELSEIF RAND:2
			PRINT 只穿着内衣的
		ENDIF
		PRINTFORML %CALLNAME:TARGET%悄悄地跑到床上、把%CALLNAME:MASTER%的頭抱起来枕在了自己的胸上
	ELSE
		PRINTFORML 早上、一醒过来就感觉身体好重……
		PRINTFORML 不知道什么时候跑到床上的%CALLNAME:TARGET%、把%CALLNAME:MASTER%当做抱枕了的様子
	ENDIF
CASE "朝口交"
	PRINTFORML 早上、一醒过来就感觉下腹部迷之舒服…
	PRINTFORML 看起来、是\@ CALCF("同居格子") ? 一起同居的 # \@%CALLNAME:TARGET%在给睡着的%CALLNAME:MASTER%口交的様子…
CASE "Ｗ朝口交"
	PRINTFORML 早上、一醒过来就感觉下腹部迷之舒服…
	PRINTFORML どうやら、%CALLNAME:PARTNER()%と%CALLNAME:TARGET%が協力して寝ている%CALLNAME:MASTER%に口交をしていたようだ…
CASE "朝六九式"
	PRINTFORML 早上、一醒过来就感觉下腹部迷之舒服…
	PRINTFORM 視線一撇…大张着双腿跨在自己身体上的%CALLNAME:TARGET%的
	IF (TALENT:淫乱 && ABL:露出癖 >= 5 && RAND:2) || CLO_ORI(TARGET,"贴身制服","胖次") == 0 || COND("身嗜み：没穿内褲")
		IF RAND:3 == 0
			PRINT 正往下流着愛液的阴部
		ELSEIF RAND:2
			PRINT 完全発情了的阴部
		ELSE
			PRINT 完全暴露在外的阴部
		ENDIF
	ELSEIF CLO_ORI(TARGET,"贴身制服")
		IF (ABL:欲望 >= 4 || TALENT:容易濕) && RAND:2
			PRINTFORM 透出水迹的胖次
		ELSEIF AGE(TARGET) >= 3
			PRINTFORM 勒入股間的胖次
		ELSE
			PRINTFORM 可愛的胖次
		ENDIF
	ELSE
		IF (ABL:欲望 >= 4 || TALENT:容易濕) && RAND:2
			PRINTFORM 透出水迹的
		ELSEIF AGE(TARGET) >= 3
			PRINTFORM 勒入股間的
		ELSE
			PRINTFORM 可愛的
		ENDIF
		IF EQUIP:胖次
			PRINTFORM %PANTIES(TARGET, "略")%
		ELSE
			PRINTFORM 胖次
		ENDIF
	ENDIF
	PRINTL 径直映入眼帘
	PRINTFORML 看起来、是%CALLNAME:TARGET%正一边给睡着的%CALLNAME:MASTER%口交、一边用股間在%CALLNAME:MASTER%脸上摩擦的様子…
ENDSELECT

;-------------------------------------------------
;朝Ｈ地の文(事後)
;-------------------------------------------------
@AFTERTEXT_NEWDAY
SELECTCASE CSTR:朝袭
CASE "朝逆正常位"
	IF TARGET == ASSI
		IF EXP_GAIN("射精経験") && RAND:2
			PRINTFORMW 完全醒来的%CALLNAME:MASTER%挺动着腰肢告诉%CALLNAME:TARGET%要好好地侍奉、然后一边享受精液叩打膣深处的触感一边允许其報告今天的日程…
		ELSE
			PRINTFORMW 確認主人醒来之后、%CALLNAME:TARGET%一边慢慢扭动着腰一边在嬌声中向%CALLNAME:MASTER%報告今天的日程…
		ENDIF
	;ＭＡＳＴＥＲ主導権な展開はまだできてない
	;ELSEIF RAND:2
	;	PRINTFORMW %CALLNAME:MASTER%が下半身の違和感に睁開眼睛と、内衣を剥ぎ取りいきり立った阴茎を突き入れようとする%CALLNAME:TARGET%と目があった
	;	PRINTFORMW しどろもどろになって言い訳するAへ你は妖艶に笑いかけ的同时、足を開いてAを誘った…
	ELSE
		IF EXP_GAIN("射精経験") && RAND:3 == 0
			PRINTFORM 在%CALLNAME:MASTER%醒来前一直阻止着射精的%CALLNAME:TARGET%、
		ELSEIF RAND:2
			PRINTFORM 通过阴道缩进的变化察觉到%CALLNAME:MASTER%醒来的%CALLNAME:TARGET%、
		ELSE
			PRINTFORM 以亲吻的方式堵住正在品味朝袭的%CALLNAME:MASTER%的嘴唇的%CALLNAME:TARGET%、
		ENDIF
		IF EXP_GAIN("射精経験") && RAND:3 == 0
			PRINTFORMW 挺动腰肢发出早上的问候的同时\@ COND("危険日", MASTER) ? 大量的精液流入了危険日的子宮 # 被大量的精液舒服地中出了 \@…
		ELSEIF ABL:腰靈活 >= 3 && RAND:3 == 0
			PRINTFORMW 巧妙地扭动着腰以提高彼此的快感的方式发出早上的问候…
		ELSEIF ABL:施虐属性 >= 3 && RAND:2
			PRINTFORMW 趁在刚睡醒\@ RAND:2 ? 抵抗弱 # 動作迟鈍 \@的时间段里\@ (EXP_GAIN("射精経験") && RAND:2) ? 更加充分榨取 # 更激烈地侵犯 \@的同时发出了早上的问候…
		ELSE
			PRINTFORMW 像敲打一样摇动着腰肢一边露出舒畅的表情一边发出了早上的问候…
		ENDIF
	ENDIF
CASE "朝舐陰"
	PRINTFORM %CALLNAME:TARGET%笑着吐了吐舌頭、
	IF ABL:奉仕精神 >= 3 && RAND:2
		PRINTFORMW 一边漂亮地对陰唇開始掃除一边进行了早上的问候…
		CALL RESET_STAIN_EX, "Ｖ", MASTER
	ELSE
		PRINTFORMW 进行了早上的问候…
	ENDIF

	;IF COND("身嗜み：没穿内褲", MASTER) == 0 && RAND:2
	;	PRINTFORMW %CALLNAME:TARGET%が部屋を出て行った後に鏡の前に立つと、無防備な下半身が目に入る
	;	PRINTFORMW …どうやら、%CALLNAME:TARGET%に胖次を持っていかれたようだ……
	;ENDIF
CASE "朝足交"
	PRINTFORM %CALLNAME:TARGET%
	IF TALENT:坏心眼 || TALENT:小悪魔
		IF EXP_GAIN("射精経験", MASTER)
			PRINTFORM 把脚上沾着的精液用手指抹了一点舔了舔、
		ELSEIF PENIS(MASTER) == 0
			PRINTFORM 把脚上沾着的愛液用手指抹了一点舔了舔、
		ENDIF
	ENDIF
	PRINTFORMW 用满是優越感的表情问了声早上好

CASE "朝の新妻PLAY"
	PRINTFORMW 不由自主地在早餐前抱了%CALLNAME:TARGET%…
	SIF TALENT:害羞 || TALENT:傲慢
		PRINTFORMW %CALLNAME:TARGET%抱怨着%CALLNAME:MASTER%的行為、看起来很高兴的样子…
CASE "朝乳交"
	IF EXP_GAIN("射精経験", MASTER)
		PRINTFORMW %CALLNAME:TARGET%用被射满了精液的脸向你问候了早安
	ELSE
		PRINTFORMW %CALLNAME:TARGET%用把脸埋进你股間的姿勢向你问候了早安
	ENDIF
CASE "朝面部騎乗"
	IF CHECK_CLO("Ｖ露出")
		PRINTFORMW 一边被%CALLNAME:TARGET%的%TEXTR("股间碾压着/股間的气味包裹着/露出来的股间肌肤清洗着脸/股間的气味熏染着")%一边睁開了眼睛
	ELSE
		PRINTFORMW 一边被%CALLNAME:TARGET%的%TEXTR("股间碾压着/股間的气味包裹着/胖次清洗着脸/股間的气味熏染着")%一边睁開了眼睛
	ENDIF
CASE "朝騎乗位"
	IF EXP_GAIN("射精経験", MASTER)
		IF TARGET != ASSI
			PRINTFORMW %CALLNAME:TARGET%一边用膣内接受不断的射精、一边进行早上的问候
		ELSE
			PRINTFORMW 一边在騎乗位的%CALLNAME:TARGET%的膣内射精、一边进行早上的问候
		ENDIF
	ELSE
		PRINTFORMW 因为騎乗位而醒了过来
	ENDIF
CASE "朝Ａ騎乗位"
	IF EXP_GAIN("射精経験", MASTER)
		IF TARGET == ASSI
			PRINTFORMW 一边在騎乗位的%CALLNAME:TARGET%的菊穴内射精、一边进行了早上的问候
		ELSE
			IF COND("Ａ清洁口交可能")
				PRINTFORML 用%TEXTR("肛門/菊穴")%%TEXTR("搾取了精液/貪求完快楽")%的%CALLNAME:TARGET%、一边%TEXTR("用舌頭清理/感謝地含进去/舔着清理")%%CALLNAME:MASTER%的汚垢阴茎、一边进行早上的问候
				;恥垢もお掃除
				CALL RESET_STAIN_EX, "Ｐ", MASTER
				PRINTFORMW 清洁口交経験＋１
				EXP:清洁口交経験 += 1
			ELSE
				PRINTFORMW %CALLNAME:TARGET%、一边用%TEXTR("肛門/菊穴/屁眼")%来%TEXTR("搾取/接受/品尝")%%CALLNAME:MASTER%的精液、一边进行早上的问候
			ENDIF
		ENDIF
	ELSE
		PRINTFORMW 因为肛門騎乗位而醒了过来
	ENDIF

CASE "朝接吻"
	IF ABL:奉仕精神 >= 5 && RANDF(2)
		PRINTFORM %CALLNAME:TARGET%
		IF RAND:4 && TALENT:害羞 && TALENT:坦率 + TALENT:相思相愛 == 0
			PRINTFORM 满脸绯红地
		ELSE
			PRINTFORM 一边抱住%CALLNAME:MASTER%\@ TALENT:好色 ? 、一边撫摸股間 #  \@、
		ENDIF
		PRINTFORMW 进行了早上的问候
	ENDIF
CASE "朝添い寝"
	PRINTFORM %CALLNAME:TARGET%
	IF RAND:4 && TALENT:害羞 && TALENT:坦率 + TALENT:相思相愛 == 0
		PRINTFORM 满脸绯红地
	ELSE
		PRINTFORM 一边抱住%CALLNAME:MASTER%\@ TALENT:好色 ? 、一边撫摸股間 #  \@、
	ENDIF
	PRINTFORMW 进行了早上的问候
CASE "朝口交", "朝六九式"
	IF (TALENT:好色 || ABL:奉仕精神 >= 5) && CLO_ORI(TARGET,"胖次") && COND("身嗜み：没穿内褲") == 0 && RAND:3 == 0
		PRINTFORM %CALLNAME:TARGET%用还残留着体温的胖次擦拭%CALLNAME:MASTER%的阴茎
		IF RAND:2
			PRINTFORMW 一边进行早上的问候
		ELSE
			PRINTFORMW 、然后在你眼前把沾满了黏液的胖次又穿回去、向你进行了早上的问候
		ENDIF
	ELSEIF CSTR:朝袭 == "朝六九式" && ABL:施虐属性 && RAND:2
		PRINTFORMW %CALLNAME:TARGET%把%CALLNAME:MASTER%的脸按在股間、隔着肩膀向你进行了早上的问候
	;清洁口交
	ELSEIF (ABL:奉仕精神 >= 5 || TALENT:献身) && RAND:3
		;ＭＡＳＴＥＲが包茎
		IF TALENT:MASTER:包茎 && ABL:奉仕精神 + TALENT:献身 >= 5 && RAND:3
			PRINTFORM %CALLNAME:TARGET%%CALLNAME:MASTER%の
			IF COND("真性包茎", MASTER)
				PRINTFORML 一边把舌頭伸进了亀頭的包皮中、将内部积攒的恥垢用舌頭全部挑出咽下去、一边进行早上的问候
			ELSE
				PRINTFORML 一边把包茎阴茎的皮剥開露出亀頭、把睡眠时間积攒的恥垢用舌頭进行掃除、一边进行早上的问候
			ENDIF
			;恥垢もお掃除
			CALL RESET_STAIN_EX, "Ｐ", MASTER
		ELSE
			IF EXP_GAIN("射精経験", MASTER)
				PRINTFORM %CALLNAME:TARGET%把精液\@ ABL:精液中毒 >= 3 ? 非常美味地 # 一点不留地 \@喝了下去、
			ELSE
				PRINTFORM %CALLNAME:TARGET%把%CALLNAME:MASTER%的阴茎
			ENDIF
			;ちんきす
			IF (TALENT:恋慕 || MARK:屈服刻印 >= 2) && RAND:2
				PRINTFORML 用舌頭掃除完之後、一边亲吻着马眼、一边进行早上的问候
			ELSE
				PRINTFORML 一边用嘴清理、一边进行早上的问候
			ENDIF
		ENDIF
		PRINTFORMW 清洁口交経験＋１
		EXP:清洁口交経験 += 1
	ELSEIF EXP_GAIN("射精経験", MASTER) && RAND:3
		IF ABL:施虐属性 >= 3 && RAND:2
			PRINTFORMW %CALLNAME:TARGET%一边享受着精液一边咽下、舔了舔嘴唇、向你进行了早上的问候
		ELSE
			PRINTFORMW %CALLNAME:TARGET%用被射满了精液的脸向你问候了早安
		ENDIF
	ELSE
		PRINTFORM %CALLNAME:TARGET%把脸埋进你的股間
		IF COND("指の使用") && RAND:4 == 0
			PRINTFORM 一边舔着手上的黏液、一边满脸笑顔地
		ELSE
			PRINTFORM %TEXTR("一边舔舐着阴茎、一边/嘴里咕隆咕隆地、用含糊不清的声音/保持着这个姿势")%
		ENDIF
		PRINTFORMW 进行了早上的问候
	ENDIF
CASE "Ｗ朝口交"
	;清洁口交
	IF (ABL:奉仕精神 >= 5 || TALENT:献身) && RAND:3
		;ＭＡＳＴＥＲが包茎
		IF TALENT:MASTER:包茎 && ABL:奉仕精神 + TALENT:献身 >= 5 && RAND:3
			PRINTFORM %CALLNAME:PARTNER()%達は%CALLNAME:MASTER%の
			IF COND("真性包茎", MASTER)
				PRINTFORML 一边把舌頭伸进了亀頭的包皮中、将内部积攒的恥垢用舌頭骚动着钻进去全部挑出咽下去、一边进行早上的问候
			ELSE
				PRINTFORML 一边把包茎阴茎的皮剥開露出亀頭、把睡眠时間积攒的恥垢用舌頭进行掃除、一边进行早上的问候
			ENDIF
			;恥垢もお掃除
			CALL RESET_STAIN_EX, "Ｐ", MASTER
		ELSE
			IF EXP_GAIN("射精経験", MASTER)
				PRINTFORM %CALLNAME:PARTNER()%们把精液\@ ABL:精液中毒 >= 3 ? 非常美味地 # 一点不留地 \@喝了下去、
			ELSE
				PRINTFORM %CALLNAME:PARTNER()%们把%CALLNAME:MASTER%的阴茎
			ENDIF
			;ちんきす
			IF (TALENT:恋慕 || MARK:屈服刻印 >= 2) && RAND:2
				PRINTFORML 用舌頭掃除完之後、一边左右亲吻着龟頭、一边进行早上的问候
			ELSE
				PRINTFORML 一边协力用嘴清理、一边进行早上的问候
			ENDIF
		ENDIF
		PRINTFORMW 清洁口交経験＋１(二人とも)
		EXP:清洁口交経験 += 1
		EXP:PARTNER():清洁口交経験 += 1
	ELSEIF EXP_GAIN("射精経験", MASTER) && RAND:3
		PRINTFORMW %CALLNAME:PARTNER()%们的脸上沾满了精液、一边左右用舌頭舔舐着阴茎一边进行早上的问候
	ELSE
		PRINTFORM %CALLNAME:PARTNER()%们将脸埋进股间
		PRINTFORM %TEXTR("一边舔舐着阴茎、一边/嘴里咕隆咕隆地、用含糊不清的声音/保持着这个姿势")%
		PRINTFORMW 进行了早上的问候
	ENDIF
ENDSELECT

@PLAY_AT_MORNING
#DIM LCOUNT

;実行可能指令の確認
CALL CHECK_COM_ABLE, TARGET

STR:選択肢 =

SIF CANTRAIN("接吻") && EXP:接吻経験 && COND("合意：接吻") && CHECK_COM("接吻") && (CFLAG:好感度 >= 1000 || TALENT:Ｍ性向 || TALENT:恋慕)
	STR:選択肢 += "朝接吻/"
;ちょっと例外なので一旦オミット
;SIF (TALENT:恋慕 || ABL:信頼 >= 5) && TALENT:母乳体質 && ABL:奉仕精神 >= 3 && EXP:噴乳経験 >= 20
;	STR:選択肢 += "朝の母乳搾り/"
SIF CANTRAIN("足交") && EXP:足交経験 && ABL:施虐属性 >= 3
	STR:選択肢 += "朝足交/"
SIF CANTRAIN("奉仕舐陰") && EXP:舐陰経験 && COND("性別嫌悪") == 0 && TALENT:MASTER:男性 == 0
	STR:選択肢 += "朝舐陰/"
SIF CANTRAIN("正常位奉仕") && PENIS(TARGET) && FIRSTCOMS("Ｖ逆挿入") && TALENT:童貞 == 0 && EXP:Ｖ挿入経験 >= 20 && ABL:欲望 >= 3 && TALENT:MASTER:処女 == 0 && TALENT:MASTER:妊娠 == 0
	STR:選択肢 += "朝逆正常位/"

IF CANTRAIN("口交") && EXP:口交経験 && PENIS(MASTER) && COND("性別嫌悪") == 0 && MAX(ABL:欲望, ABL:奉仕精神) >= 3 && (ABL:精液中毒 >= 1 || ABL:奉仕精神 >= 5)
	CALL COM_ABLE99
	IF CANTRAIN("六九式") && PENIS(TARGET) == 0 && COND("脚の使用") && ABL:欲望 >= 3 && ABL:奉仕精神 >= 3 && CLO_ORI(TARGET,"外套") == 0 && RAND:2
		STR:選択肢 += "朝六九式/"
	ELSEIF COND("Ｗ朝口交可能") && FIND_COND("Ｗ朝口交可能", "相性○") && RAND:2
		STR:選択肢 += "Ｗ朝口交/"
	ELSE
		STR:選択肢 += "朝口交/"
	ENDIF
ENDIF

SIF CANTRAIN("推倒舐陰") && PENIS(TARGET) == 0 && CHECK_CLO("胖次露出") && ABL:施虐属性 && (ABL:施虐属性 >= 3 || ABL:欲望 >= 5) && ABL:Ｃ感覚 >= 3
	STR:選択肢 += "朝面部騎乗/"
SIF CANTRAIN("騎乗位") && EXP:Ｖ性交経験 && PENIS(MASTER) && (ABL:奉仕精神 >= 5 || (ABL:欲望 >= 4 && ABL:Ｖ感覚 >= 3) ) && TALENT:処女 + TALENT:男性 + TALENT:PLAYER:童貞 == 0 && COND("女性主導生性交")
	STR:選択肢 += "朝騎乗位/"
SIF CANTRAIN("肛門騎乗位") && COND("Ａ熟達") > COND("Ｖ熟達") && PENIS(MASTER) && (ABL:奉仕精神 >= 5 || (ABL:欲望 >= 4 && ABL:Ａ感覚 >= 3) ) && EXP:PLAYER:Ａ挿入経験
	STR:選択肢 += "朝Ａ騎乗位/"
SIF CANTRAIN("乳交") && EXP:乳交経験 && PENIS(MASTER) && COND("性別嫌悪") == 0 && ABL:奉仕精神 >= 4 && (COND("伪娘") || TALENT:男性 == 0)
	STR:選択肢 += "朝乳交/"
IF CANTRAIN("後背位") && PENIS(MASTER) && TALENT:恋慕
	IF TALENT:男性 == 0
		SIF EXP:Ｖ性交経験 && TALENT:処女 + TALENT:PLAYER:童貞 == 0 && COND("女性主導生性交")
			STR:選択肢 += "朝の新妻PLAY/"
	ELSEIF COND("伪娘")
		SIF EXP:Ａ性交経験 && FIRSTCOMS("Ａ挿入") && COND("Ａ熟達")
			STR:選択肢 += "朝の新妻PLAY/"
	ENDIF
ENDIF

IF STR:選択肢 == ""
	STR:選択肢 += "朝添い寝/"
ELSEIF COND("夏") == 0 && (ABL:信頼 >= 5 || CFLAG:好感度 >= 2000 || TALENT:恋慕)
	STR:選択肢 += "朝添い寝/"
ENDIF

CSTR:朝袭 = %TEXTR(STR:選択肢)%

RETURN 0

license: unknown


code:
;---------------------------------------------------------
;妊娠イベントカウント処理
;---------------------------------------------------------
@EVENT_N1, ARG
#DIM FATHER
LOCAL = TARGET
TARGET = ARG

CALL CHECK_NINSHIN, TARGET, "システム処理"

;妊娠イベントカウント処理
;妊娠事件计数处理
IF COND("夜")
	;***着床判定追加パッチで追加
	;***通过着床判定追加补丁追加
	IF BASE:ARG:受精 > 0 && COND("精液の優勢")
		;受精からの経過日数カウントアップ
		;受精后经过的天数增加
		BASE:ARG:受精からの日数 += 1
		;十分な日数が経っていたら着床判定(４～６日は長いので１日短縮)
		;如果过了足够的天数，则判定为着床（4~6天很长，所以缩短了1天）
		IF BASE:ARG:受精からの日数 >= RAND(3, 6)
			IF COND("着床確率")
				;着床判定
				DRAWLINE
				DRAWLINE
				PRINTFORM %CALLNAME:ARG%的受精卵准备着床了…
				CALL PRINT_STRW, @" (着床確率:_黄色_{COND("着床確率")}％)"
				CALL PRINT_STRW, @"%CALLNAME:ARG%的子宮内的%CONDS("精液の優勢", ARG)%和%CALLNAME:ARG%的受精卵…"
				;受精卵が子宮に着床するか判定…
				IF RAND:100 <= COND("着床確率")
					;おめでた
					BASE:ARG:胎児 = 1
					;***種付け強化パッチで追加
					;父親の孕ませ経験UP
					FATHER = COND("精液の優勢")
					;IF (EXP:FATHER >= 0 )
					IF FATHER >= 0
						EXP:FATHER:孕ませ経験 += 1
					ENDIF
					;孕ませ素質の追加は妊娠発覚時に行う
					;20%で息子（産まれる時に選べるようにもしてあるので実はなくても良い）
					SIF RAND:5 == 0
						CALL SETFLAG, "妊娠中：息子", ARG
					CALL PRINT_STRW, "平安地着床了…_H"
					IF FATHER >= 0
						PRINTFORMW %CONDS("精液の優勢")%の孕ませ経験+１ (現在：{EXP:FATHER:孕ませ経験})
					ELSE
						PRINTFORMW %CONDS("精液の優勢")%の孕ませ経験+１
					ENDIF
				ELSE
					PRINTFORMW 没有着床
				ENDIF
			ENDIF

			;着床確実フラグOFF(事后避孕药などで着床を阻止された場合は、ここでOFFになり効果を失う)
			CFLAG:ARG:着床確実 = 0
			;受精状態は解除
			BASE:ARG:受精 = 0
			BASE:ARG:受精からの日数 = 0
		ENDIF
	;受精状態じゃない時には着床確実フラグをOFFにしておく
	ELSE
		CFLAG:ARG:着床確実 = 0
	ENDIF

	;事后避孕药は徐々に効果を失う
	SIF TALENT:ARG:事后避孕药 > 0
		TALENT:ARG:事后避孕药 -= 1

	;お腹の中の胎児の成長
	IF COND("妊娠中")
		;すくすく育ってます
		BASE:胎児 += 1
		;中出しによって速度アップ
		SIF RAND:(3 + BASE:胎児への栄養/10)
			BASE:胎児 += 1

		;分娩室待ち
		IF COND("妊娠中：分娩直前")
			CALL SETFLAG, "妊娠中：分娩", TARGET
		;臨月～出産日(人なら三日ほど)
		ELSEIF COND("妊娠中：臨月")
			;出産間近イベント(子育て中には発生しない)
			IF COND("妊娠中：各イベント進行度") >= 3
				IF BASE:育児中
					BASE:胎児 -= 1 + RAND:2
				ELSE
					CALL SETFLAG, "妊娠中：分娩直前", TARGET
				ENDIF
			ENDIF
		;妊娠発覚終了～臨月まで(人なら10日ほど)
		ELSEIF COND("妊娠中：妊娠発覚済み")
			;臨月イベント
			SIF COND("妊娠中：各イベント進行度") >= 11
				CALL SETFLAG, "妊娠中：臨月", TARGET
		;妊娠発覚イベント待ち
		ELSEIF COND("妊娠中：妊娠発覚直前")

		;妊娠潜伏期間～妊娠発覚イベント(人なら三日ほど)
		ELSEIF COND("妊娠中：各イベント進行度") >= 4
			CALL SETFLAG, "妊娠中：妊娠発覚直前", TARGET
		ENDIF
	ENDIF
	;子育て。人で２週間程度。
	IF BASE:育児中
		BASE:育児中 += 1
		;子育て終了
		SIF COND("育児中：兒童が大きくなった")
			CALL SETFLAG, "育児中：育児終了", TARGET
	ENDIF
ENDIF

TARGET = LOCAL

;---------------------------------------------------------
;妊娠判定処理
;ARGSには"システム処理"が入る。これ以外の場合は時間経過によるシステム処理以外での妊娠判定となる
;CHECK_NINSHIN_ALLACTORは全部のPLAYER:0～1とTARGET:0～3について行う
;---------------------------------------------------------
@CHECK_NINSHIN_ALLACTOR, ARGS
#DIM LCOUNT

FOR LCOUNT, 1, CHARANUM
	SELECTCASE LCOUNT
	CASE PLAYER, PLAYER:1, TARGET, TARGET:1, TARGET:2, TARGET:3
		CALL CHECK_NINSHIN, LCOUNT, ARGS
	ENDSELECT
NEXT

@CHECK_NINSHIN, ARG, ARGS
#DIM LCOUNT
#DIM MEMO_LINECOUNT

SIF ARG < 0
	RETURN 0
SIF ARG == 0 && TARGET
	ARG = TARGET

SELECTCASE ARGS
CASE "システム処理"
	;PRINTFORM [{ARG, 2}]%TEXT_LJ(CALLNAME:ARG, 16)% {DAY - CFLAG:ARG:最新中出し日, 3}日前, 妊娠率：{COND("受精確率"), 3}％, {COND("胎内の精液量", ARG), 6}/{BASE:ARG:胎内精子量, 4} →

	;中出し時に既に判定行っているので精子量の減少処理を行う
	CALL CHECK_NINSHIN_SPERM, ARG

	IF TALENT:ARG:Ａ妊娠体質 == 0
		;***着床判定追加パッチで変更
		;妊娠判定(出産後すぐに妊娠できるようにしてみる)　(着床判定有効時は、受精判定)
		IF BASE:ARG:胎内精子量 && BASE:ARG:受精 == 0 && BASE:ARG:胎児 == 0 && COND("Ｖ妊娠可能", ARG)
			IF CALCF("妊娠判定", ARG)
				;受精
				BASE:ARG:受精 = CALCF("受精強度算出", ARG)
				BASE:ARG:受精からの日数 = 0
				;超危険日ならば着床確実フラグON
				SIF COND("排卵中", ARG)
					CFLAG:ARG:着床確実 = 1
			ENDIF
		ENDIF
	ELSE
		IF BASE:ARG:Ａ胎内精子量 && BASE:ARG:受精 == 0 && BASE:ARG:胎児 == 0 && COND("Ａ妊娠可能", ARG)
			IF CALCF("妊娠判定", ARG)
				;受精
				BASE:ARG:受精 = CALCF("受精強度算出", ARG)
				BASE:ARG:受精からの日数 = 0
				;超危険日ならば着床確実フラグON
				SIF COND("排卵中", ARG)
					CFLAG:ARG:着床確実 = 1
			ENDIF
		ENDIF
	ENDIF

	;創口貼装着日数カウント
	SIF CFLAG:ARG:創口貼装着
		CFLAG:ARG:創口貼装着 += 1
	;5日で剥がれる
	IF CFLAG:ARG:創口貼装着 > 9
		PRINTFORML
		DRAWLINE
		PRINTFORML %CALLNAME:ARG%的創口貼由于时间而剥落了…
		CFLAG:ARG:創口貼装着 = 0
		;とりあえず半分に
		SIF BASE:ARG:胎内精液量
			BASE:ARG:胎内精液量 /= 2
		SIF BASE:ARG:胎内精子量
			BASE:ARG:胎内精子量 /= 2
	ENDIF

	;PRINTFORML 妊娠率：{COND("受精確率"), 3}％, {COND("胎内の精液量", ARG), 6}/{BASE:ARG:胎内精子量, 4}
CASEELSE
	;受精確率が変動しているかを格子
	SIF COND("受精確率", ARG) == TCVAR:ARG:受精確率
		RETURN 0

	;受精確率の変動を表示
	IF CONFIG("妊娠設定：受精確率表示ＯＦＦ") == 0
		LOCALS = %CALLNAME:ARG%の受精確率：_

		;最終確率
		SELECTCASE COND("受精確率", ARG)
		CASE IS >= 100
			;***着床判定追加パッチで変更
			LOCALS = %LOCALS%爱心粉紅_受精確実！_
		CASE IS > 50
			LOCALS = %LOCALS%爱心粉紅_{COND("受精確率", ARG)}％_
		CASE IS > 20
			LOCALS = %LOCALS%黄色_{COND("受精確率", ARG)}％_
		CASEELSE
			LOCALS = %LOCALS%{COND("受精確率", ARG)}％_
		ENDSELECT

		;伸び幅
		SELECTCASE COND("受精確率", ARG) - TCVAR:ARG:受精確率
		CASE IS >= 15
			LOCALS = %LOCALS%粉紅_（↑{COND("受精確率", ARG) - TCVAR:ARG:受精確率}％）_
		CASE IS >= 10
			LOCALS = %LOCALS%爱心粉紅_（↑{COND("受精確率", ARG) - TCVAR:ARG:受精確率}％）_
		CASE IS >= 5
			LOCALS = %LOCALS%黄色_（↑{COND("受精確率", ARG) - TCVAR:ARG:受精確率}％）_
		CASE IS > 0
			LOCALS = %LOCALS%（↑{COND("受精確率", ARG) - TCVAR:ARG:受精確率}％）_
		CASE IS < 0
			LOCALS = %LOCALS%青_（↓{TCVAR:ARG:受精確率 - COND("受精確率", ARG)}％）_
		ENDSELECT

		CALL PRINT_STRL, LOCALS
		PRINTL
	ENDIF

	;ここで妊娠判定も行ってしまう
	IF CALCF("妊娠判定", ARG)
		;***着床判定追加パッチで変更
		;淫紋ビフォー
		IF MARK:ARG:淫紋
			CALL PRINT_CUTIN, "淫紋", ARG
			PRINTFORMW %CALLNAME:ARG%的淫紋的様子……颜色和形状不停变化着！
		ENDIF

		;受精
		BASE:ARG:受精 = CALCF("受精強度算出", ARG)
		BASE:ARG:受精からの日数 = 0
		;超危険日ならば着床確実フラグON
		SIF COND("排卵中", ARG)
			CFLAG:ARG:着床確実 = 1

		CALL SETFLAG, "受精確認フラグ", ARG

		;淫紋アフター
		IF MARK:ARG:淫紋
			CALL CLEAR_CUTIN
			CLEARLINE 1
			CALL PRINT_CUTIN, "淫紋", ARG
			PRINTFORMW %CALLNAME:ARG%的淫紋的様子……颜色和形状不停变化着！
		ENDIF

		CALL PRINT_CUTIN, "受精", TARGET
		CALL PRINT_STRFW, @"UH_%CALLNAME:ARG%被%CONDS("精液の優勢", ARG)%的精子受精了_UH"
		PRINTL

		IF MARK:ARG:淫紋
			;セリフがあるかの判定
			MEMO_LINECOUNT = LINECOUNT

			CALL KOJO_GUEST, "受精確認", ARG

			SIF LINECOUNT > MEMO_LINECOUNT
				PRINTL
		ENDIF
	ENDIF

	TCVAR:ARG:受精確率 = COND("受精確率", ARG)

ENDSELECT

;---------------------------------------------------------
;胎内の精液量＆精子量を減少させる。だいたい３日程で死滅。
;ちなみにCFLAG:最新中出し日は最新の中出し日
;Ａ妊娠体質は通常のＶ妊娠とは排他の関係にある
;---------------------------------------------------------
@CHECK_NINSHIN_SPERM, ARG
#DIM TIME_PASS
;DEC_VALUE:0は精液の減少レートで、DEC_VALUE:1は精子の減少レート
#DIM DEC_VALUE, 2
VARSET DEC_VALUE

IF TALENT:ARG:Ａ妊娠体質
	TIME_PASS = DAY - CFLAG:ARG:Ａ最新中出し日
	BASE:ARG:胎内精液量 = 0
	BASE:ARG:胎内精子量 = 0
ELSE
	TIME_PASS = DAY - CFLAG:ARG:最新中出し日
	BASE:ARG:Ａ胎内精液量 = 0
	BASE:ARG:Ａ胎内精子量 = 0
ENDIF

SELECTCASE TIME_PASS
;出した直後
CASE 0
	DEC_VALUE:0 = 120 + TALENT:ARG:吸精体質*10 + MARK:ARG:淫紋*5 + 10*RAND:6
	DEC_VALUE:1 = 100 + TALENT:ARG:吸精体質*10 + MARK:ARG:淫紋*5 + 10*RAND:6
CASE 1
	DEC_VALUE:0 = 150 + TALENT:ARG:吸精体質*10 + MARK:ARG:淫紋*5 + 10*RAND:11
	DEC_VALUE:1 = 100 + TALENT:ARG:吸精体質*10 + MARK:ARG:淫紋*5 + 10*RAND:6
CASE 2
	SIF COND("昼")
		DEC_VALUE:0 = 200 + TALENT:ARG:吸精体質*10 + MARK:ARG:淫紋*5 + 10*RAND:11
	DEC_VALUE:1 = 100 + TALENT:ARG:吸精体質*10 + MARK:ARG:淫紋*5 + 10*RAND:6
CASE IS >= 3
	DEC_VALUE:0 = 200 + TALENT:ARG:吸精体質*10 + MARK:ARG:淫紋*5 + 10*RAND:11
	SIF COND("昼")
		DEC_VALUE:1 = 100 + TALENT:ARG:吸精体質*10 + MARK:ARG:淫紋*5 + 10*RAND:6
ENDSELECT

IF TALENT:ARG:Ａ妊娠体質
	;肛塞没有だと精液が減少
	SIF TALENT:ARG:常時Ａ肛塞 == 0 && DEC_VALUE:0
		BASE:ARG:Ａ胎内精液量 = DIVIDE(BASE:ARG:Ａ胎内精液量, DEC_VALUE:0)

	IF CONFIG("胎内精子で妊娠") && DEC_VALUE:1
		BASE:ARG:Ａ胎内精子量 = DIVIDE(BASE:ARG:Ａ胎内精子量, DEC_VALUE:1)
	ELSE
		BASE:ARG:Ａ胎内精子量 = 0
	ENDIF
	;精子が死滅したら父親フラグは消滅
	;***着床判定追加パッチで変更
	SIF BASE:ARG:Ａ胎内精子量 == 0 && BASE:ARG:胎児 == 0 && BASE:ARG:受精 == 0
		BASE:ARG:精液の優勢 = 0
ELSE
	;創口貼没有だと精液が減少
	SIF CFLAG:ARG:創口貼装着 == 0 && DEC_VALUE:0
		BASE:ARG:胎内精液量 = DIVIDE(BASE:ARG:胎内精液量, DEC_VALUE:0)
	;胎内精子で妊娠がONでも精子は減っていく　4日程で死滅
	IF CONFIG("胎内精子で妊娠") && DEC_VALUE:1
		BASE:ARG:胎内精子量 = DIVIDE(BASE:ARG:胎内精子量, DEC_VALUE:1)
	ELSE
		BASE:ARG:胎内精子量 = 0
	ENDIF
	;精子が死滅したら父親フラグは消滅
	;***着床判定追加パッチで変更
	SIF BASE:ARG:胎内精子量 == 0 && BASE:ARG:胎児 == 0 && BASE:ARG:受精 == 0
		BASE:ARG:精液の優勢 = 0
ENDIF

;---------------------------------------------------------
;妊娠発覚＆出産予定報告＆子の親出口処理
;---------------------------------------------------------
@EVENT_N2, ARG
#DIM MEMO_TARGET
MEMO_TARGET = TARGET

TARGET = ARG

;子育て中
IF BASE:育児中
	;子の親出口
	IF COND("育児中：育児終了")
		DRAWLINE
		;PRINTFORMW %CALLNAME:TARGET%の育てていた兒童は館に充満する力のせいか、かなり早く大きくなりました。
		PRINTFORMW %CALLNAME:TARGET%养育的孩子、很快就长大了。
		PRINTFORMW 已经不需要再照顾了。

		CALL KOJO_EVENT, "子出口"

		BASE:育児中 = 0
	ENDIF
ENDIF

;出産予定報告(子育て中には産まれない)
IF COND("妊娠中：分娩直前") && TALENT:妊娠 == 2 && BASE:育児中 == 0
	DRAWLINE
	PRINTFORMW 在%CALLNAME:TARGET%的腹中的胎儿活动得很活跃。
	PRINTFORMW 好像明日就要出産的样子…
	TALENT:妊娠 = 3
ELSEIF COND("妊娠中：臨月") && TALENT:妊娠 == 1
	DRAWLINE
	PRINTFORMW %CALLNAME:TARGET%的肚子已经变大得很明显了。
	PRINTFORMW 好像再过几天就能生産了…
	TALENT:妊娠 = 2
;子育て中以外の妊娠発覚
ELSEIF COND("妊娠中：妊娠発覚直前") && TALENT:妊娠 == 0 && COND("妊娠中：各イベント進行度") >= 2
	TALENT:妊娠 = 1
	;妊娠確定フラグに変更
	CALL SETFLAG, "妊娠中：妊娠発覚済み", TARGET

	CALL EVENT_N2_TEXT

	IF COND("妊娠中") && CFLAG:妊娠で巨乳化 == 0
		PRINTL
		PRINTFORMW 伴随着妊娠、%CALLNAME:TARGET%的胸部膨胀起来了。
		;素質変更など
		IF TALENT:母乳体質 == 0
			TALENT:母乳体質 = 2
			SETBIT CFLAG:妊娠で巨乳化, 2
		ENDIF
		;胸の尺寸
		IF TALENT:男性
			PRINTFORMW 但是、好像变化不大的样子…
			PRINTFORMW %CALLNAME:TARGET%獲得了[母乳体質]。
		ELSEIF TALENT:貧乳
			SETBIT CFLAG:妊娠で巨乳化, 0
			;２階級特進
			IF RAND:3 == 0
				PRINTFORMW 这种事简直是不思議、变得\@ TALENT:娇小 ? 超出%CALLNAME:MASTER%预想般的 # \@巨大…
				PRINTFORMW %CALLNAME:TARGET%獲得了\@ GETBIT(CFLAG:妊娠で巨乳化, 2) ? [母乳体質]和 # \@[巨乳]。
				CALL CALC, "巨乳化", TARGET
			ELSE
				PRINTFORMW 看了情况之后、只是变到了普通程度的样子…
				PRINTFORMW %CALLNAME:TARGET%\@ GETBIT(CFLAG:妊娠で巨乳化, 2) ? 得到了[母乳体質]、 # \@失去了[貧乳]。
				CALL CALC, "並乳化", TARGET
			ENDIF
			PRINTFORMW
		ELSEIF TALENT:貧乳 == 0 && TALENT:巨乳 == 0
			SETBIT CFLAG:妊娠で巨乳化, 1
			PRINTFORMW 很显然、变得相当大了…
			PRINTFORMW %CALLNAME:TARGET%獲得了\@ GETBIT(CFLAG:妊娠で巨乳化, 2) ? [母乳体質]和 # \@[巨乳]。
			CALL CALC, "巨乳化", TARGET
		ELSEIF GETBIT(CFLAG:妊娠で巨乳化, 2)
			PRINTFORMW %CALLNAME:TARGET%獲得了[母乳体質]。
		ENDIF
	ENDIF
ENDIF

;妊娠に付随する母乳体質や巨乳の消去
IF CFLAG:妊娠で巨乳化 && COND("妊娠中") == 0 && BASE:育児中 == 0
	PRINTL
	PRINTFORMW （完成了任务的%CALLNAME:TARGET%的胸回归了原来的大小）
	SIF GETBIT(CFLAG:妊娠で巨乳化, 0)
		CALL CALC, "貧乳化", TARGET
	SIF GETBIT(CFLAG:妊娠で巨乳化, 1)
		CALL CALC, "並乳化", TARGET
	SIF GETBIT(CFLAG:妊娠で巨乳化, 2)
		TALENT:母乳体質 = 0

	CFLAG:妊娠で巨乳化 = 0
ENDIF

TARGET = MEMO_TARGET

;---------------------------------------------------------
;妊娠発覚イベント
;CALLNAME:(COND("精液の優勢") ) とかなると見苦しいので、FATHER = COND("精液の優勢")とする
;---------------------------------------------------------
@EVENT_N2_TEXT
#DIM FATHER

FATHER = COND("精液の優勢")

DRAWLINE

IF TARGET == MASTER
	PRINTFORMW %CALLNAME:MASTER%的身体不太舒服、稍微仔细检查了一下…
	PRINTFORMW ………
	PRINTFORMW ……
	PRINTFORMW …
	PRINTFORMW 調查的結果、\@ BASE:育児中 ? 在养育孩子的途中 # 很显然 \@
	SELECTCASE FATHER
	CASE IS > 0
		PRINTFORM %CALLNAME:FATHER%の
	ENDSELECT
	PRINTFORML 孩子好像怀上了。
;子育て中
ELSEIF BASE:育児中
	PRINTFORMW %CALLNAME:TARGET%诉说着身体的变化…
	PRINTFORML 調查的結果、%CALLNAME:TARGET%在养育孩子的途中再次妊娠了。
;恋慕
ELSEIF TALENT:恋慕
	PRINTFORMW %CALLNAME:TARGET%将妊娠的事说了出来…
	PRINTFORML 調查的結果、%CALLNAME:TARGET%确实妊娠了。
;其他
ELSE
	PRINTFORMW %CALLNAME:TARGET%诉说着身体的变化…
	PRINTFORML 調查的結果、%CALLNAME:TARGET%妊娠了。
ENDIF

FORCEWAIT

CALL KOJO_EVENT, "妊娠発覚"

;MASTER以外はリアクション
IF TARGET == MASTER

;恋慕
ELSEIF TALENT:恋慕
	;父親はMASTER
	IF FATHER == MASTER
		;叛逆or自尊心高いor強気
		IF TALENT:叛逆 || TALENT:高慢 || TALENT:強気
			PRINTFORMW %CALLNAME:TARGET%露出鼓起来的肚子、索求着抚摸…
		;坦率or楽観的
		ELSEIF TALENT:坦率 || TALENT:開朗
			PRINTFORMW %CALLNAME:TARGET%微笑看着%CALLNAME:MASTER%、撫摸着鼓起来的肚子…
		ELSE
			PRINTFORMW %CALLNAME:TARGET%撫摸着鼓起来的肚子、笑得很開心…
		ENDIF
	ELSE
		PRINTFORMW %CALLNAME:TARGET%对于怀上了不是%CALLNAME:MASTER%的孩子这件事很遗憾的様子…
	ENDIF
;子育て中
ELSEIF BASE:育児中
	PRINTFORMW %CALLNAME:TARGET%一副难以言喻的表情…
;叛逆
ELSEIF TALENT:叛逆
	;出産経験有
	IF TALENT:出産経験
		PRINTFORMW %CALLNAME:TARGET%带着放弃了什么的微笑、撫摸着鼓起来的肚子…
	;出産経験无
	;叛逆or自尊心高いor強気
	ELSEIF TALENT:叛逆 || TALENT:高慢 || TALENT:強気
		PRINTFORMW %CALLNAME:TARGET%温柔地撫摸着鼓起来的肚子、但在%CALLNAME:MASTER%靠近的时候却狠狠地瞪着眼睛…
	ELSEIF TALENT:内気
		PRINTFORML 听到了这件事的%CALLNAME:TARGET%脸色发青、捂住了嘴但下一个瞬间就嘔吐了起来、
		PRINTFORMW 吐瀉物洒落在床上…
	;一線or保守的or堅強
	ELSEIF TALENT:難以越過的底線 || TALENT:保守的 || TALENT:堅強
		PRINTFORMW %CALLNAME:TARGET%用空洞的眼神看着鼓起来的肚子…
	ELSE
		PRINTFORMW 听到了这件事的%CALLNAME:TARGET%嗚咽了起来…
	ENDIF
;屈服刻印Lv2or信頼4LV
ELSEIF MARK:屈服刻印 >= 2 || ABL:信頼 >= 4
	;叛逆or自尊心高いor強気
	IF TALENT:叛逆 || TALENT:高慢 || TALENT:強気
		PRINTFORMW %CALLNAME:TARGET%露出了複雑的表情、但还是撫摸着鼓起来的肚子…
	ELSEIF TALENT:内気
		PRINTFORMW 听到了这件事的%CALLNAME:TARGET%非常不安、悄悄地握住了%CALLNAME:MASTER%的手…
	ELSE
		PRINTFORMW %CALLNAME:TARGET%用充满愛情的手抚摸着自己的肚子…
	ENDIF
;其他
ELSEIF FATHER > 0
	PRINTFORMW %CALLNAME:TARGET%对于和%CALLNAME:FATHER%之間有了孩子这件事很震惊…
ELSE
	PRINTFORMW %CALLNAME:TARGET%对于和有了孩子这件事很震惊…
ENDIF

;意外的妊娠の場合は特別経験Get
CALL RECORD_SP_EXP

;***種付け強化パッチで追加
;父親は孕ませ経験により素質獲得
IF FATHER > 0
	IF EXP:FATHER:孕ませ経験 >= 30
		SIF TALENT:FATHER:小种马 < 2
			CALL SETFLAG, "大种马", FATHER
	ELSEIF EXP:FATHER:孕ませ経験 >= 10
		SIF TALENT:FATHER:小种马 == 0
			CALL SETFLAG, "小种马", FATHER
	ENDIF
ENDIF
;---------------------------------------------------------
;其他の妊娠イベント処理
;CALLNAME:(COND("精液の優勢") ) とかなると見苦しいので、FATHER に代入する
;---------------------------------------------------------
@EVENT_N3, ARG
#DIM FATHER
#DIM CHOICE

LOCAL = TARGET
TARGET = ARG

;お産(子育て中には出産しない)
IF COND("妊娠中：分娩") && TALENT:妊娠 && BASE:育児中 == 0
	FATHER = COND("精液の優勢")
	DRAWLINE
	IF TARGET == MASTER
		PRINTFORM %CALLNAME:MASTER%
		SELECTCASE FATHER
		CASE IS > 0
			PRINTFORM %CALLNAME:FATHER%
			SIF TALENT:出産経験
				PRINTFORM と兒童
		CASEELSE
			SIF TALENT:出産経験
				PRINTFORM 自己的孩子
		ENDSELECT
		SIF CFLAG:息子の数 + CFLAG:娘の数 >= 2
			PRINT 们
		SIF FATHER > 0 || TALENT:出産経験
			PRINT 的守望中
		PRINT 、
	ELSE
		PRINTFORMW %CALLNAME:TARGET%快要生了。
		PRINTFORMW ………
		PRINTFORMW ……
		PRINTFORMW …
		PRINTFORMW

		PRINTFORM %CALLNAME:TARGET%
		IF TALENT:子宮性感 && TALENT:淫壷
			PRINTFORMW 在出產的淫悅中逐漸放鬆了表情
		ELSEIF TALENT:出産経験 >= 3
			PRINTFORM 幾乎沒有怎麼出汗的樣子
		ELSE
			PRINTFORM 大量的汗水佈滿了臉頰
		ENDIF
		PRINTFORMW 打開大腿。
		;卵生の場合(まだ未実装)
		IF 0
			PRINTFORMW 在%CALLNAME:TARGET%不穩的喘息聲中、\@ TALENT:Ａ妊娠体質 ? 肛門 # 膣口 \@緩緩地擴張開……
			IF TALENT:Ａ妊娠体質
				PRINTFORML 本來作為排泄孔的%CALLNAME:TARGET%的肛門中出現了一個白色的物體、
			ELSE
				PRINTFORML %CALLNAME:TARGET%膣口大大地張開之後、其中一個白色的物體緩緩出現在了視野中、
			ENDIF
			PRINTFORMW 那――和%CALLNAME:TARGET%的愛的結晶帶著渾濁的聲響從%CALLNAME:TARGET%的身體中滑落出來、光滑的表面反射著妖艷的光澤展現在%CALLNAME:MASTER%的眼前。
		ELSE
			PRINTFORMW 在%CALLNAME:TARGET%不穩的喘息聲中、\@ TALENT:Ａ妊娠体質 ? 肛門 # 膣口 \@緩緩地擴張開……
			PRINTFORM 最終在深處出現了一個
			SELECTCASE CONDS("肤色")
			CASE "褐色肌"
				PRINTFORM 褐色的
			CASE "色白"
				PRINTFORM 純白的
			CASE "青肌"
				PRINTFORM 像是冬季的夜空一樣蒼青色的
			CASE "普通"
				PRINTFORM 橙紅色的
			ENDSELECT
			PRINTFORMW 頭、新的生命以這種姿態緩緩降生於這個世界。
		ENDIF

		PRINTL
		PRINTFORM %CALLNAME:TARGET%%CALLNAME:MASTER%
		SIF TALENT:出産経験
			PRINT 和兒童
		SIF CFLAG:息子の数 + CFLAG:娘の数 >= 2
			PRINT 们
		PRINTFORM 一边鼓励着、
	ENDIF
	PRINTFORMW 平安地出産了。

	;試しに兒童の性別選ばせてみるぜ
	PRINTL
	PRINTFORML 出生的孩子……
	PRINTFORML [ 0] 女孩子
	PRINTFORML [ 1] 男孩子
	CALL INPUT_SELECT, 2
	CHOICE = RESULT

	CALL BIRTH_RECORD, "出産記録：保存", TARGET, CHOICE

	IF CHOICE == 0
		CLEARBIT BASE:胎児, 20
	ELSE
		CALL SETFLAG, "妊娠中：息子", TARGET
	ENDIF

	IF COND("妊娠中：息子")
		PRINTFORMW \@ FATHER > 0 ? 长得和父親很像的 # \@元気的男孩子。
		CFLAG:息子の数 += 1
		;息子フラグ
		CALL SETFLAG, "育児中：息子", TARGET
	ELSE
		PRINTFORMW \@ TARGET == MASTER ? %CALLNAME:MASTER% # 和母親 \@很像的可愛的女孩子。
		CFLAG:娘の数 += 1
	ENDIF

	;恋慕
	IF TALENT:恋慕
		PRINTFORMW 抱着婴儿的%CALLNAME:TARGET%、让%CALLNAME:MASTER%也抱一抱这个孩子。
	ELSEIF TALENT:叛逆
		PRINTFORMW 抱着婴儿的%CALLNAME:TARGET%、正对孩子微笑着。
	ELSEIF TARGET != MASTER
		PRINTFORMW 抱着婴儿的%CALLNAME:TARGET%、朝孩子和%CALLNAME:MASTER%微笑着。
	ENDIF

	CALL KOJO_EVENT, "出産"

	TARGET = ARG

	;妊娠＆胎児フラグ終了
	BASE:胎児 = 0
	TALENT:妊娠 = 0

	;創口貼を装備していた場合は取下
	IF CFLAG:創口貼装着
		CFLAG:創口貼装着 = 0
		BASE:胎内精液量 = 0
		BASE:胎内精子量 = 0
	ENDIF

	;子育て開始
	BASE:育児中 += 1
	;兒童の数+1
	TALENT:出産経験 += 1
	;父親フラグの確定
	BASE:胎児の父親 = BASE:精液の優勢
	BASE:精液の優勢 = 0

	;一時的な素質の消去(まだ消えません)
	IF CFLAG:妊娠で巨乳化
		PRINTL
		PRINTFORMW （出産结束了、%CALLNAME:TARGET%的胸恢复了原来的样子…）
	ENDIF
	SETBIT CFLAG:MASTER:行動, 0
;子育て中
ELSEIF BASE:育児中 && RAND:4 && COND("手一杯", MASTER) + COND("同衾イベント", MASTER) == 0
	FATHER = COND("胎児の父親")
	DRAWLINE
	;妊娠しているのはMASTER
	IF TARGET == MASTER
		PRINTFORMW %CALLNAME:MASTER%忙着照顾孩子…
		SETBIT CFLAG:MASTER:行動, 0
	;同居していない
	ELSEIF COND("同居中") == 0
		CLEARLINE 1
	;父親はMASTERor既にいないor恋慕
	ELSEIF FATHER == MASTER || FATHER < 0 || TALENT:恋慕
		PRINTFORMW %CALLNAME:MASTER%去看%CALLNAME:TARGET%和孩子的情况。
		IF TALENT:叛逆
			IF RAND:5 == 0
				PRINTFORMW 进不去房间…
			ELSEIF RAND:4 == 0
				PRINTFORMW %CALLNAME:TARGET%紧紧盯着%CALLNAME:MASTER%…
			ELSEIF RAND:3 == 0
				PRINTFORMW %CALLNAME:TARGET%一看到%CALLNAME:MASTER%的脸就像要保护孩子一样紧紧地抱住…
			ELSEIF RAND:2 == 0
				PRINTFORMW %CALLNAME:TARGET%開心地和孩子玩耍。
				PRINTFORMW %CALLNAME:MASTER%就这样回到了自己的房间…
			ELSE
				PRINTFORMW %CALLNAME:TARGET%欢声笑语地和孩子玩耍、
				PRINTFORMW 但在%CALLNAME:MASTER%進入房間时便停止了…
			ENDIF
		ELSEIF RAND:6 == 0
			PRINTFORMW %CALLNAME:TARGET%正在给孩子喂奶。
			PRINTFORMW %CALLNAME:MASTER%看着%CALLNAME:TARGET%作为母親的样子、放松了下来…
		ELSEIF RAND:5 == 0
			PRINTFORMW 二人睡意朦胧地来迎接了、所以决定今夜一起睡觉…
			SETBIT CFLAG:MASTER:行動, 0
			CFLAG:MASTER:一緒に寝た = TARGET
			CFLAG:一緒に寝た = MASTER
		ELSEIF RAND:4 == 0
			PRINTFORMW 看到%CALLNAME:TARGET%正为睡不着觉的孩子发愁、决定一起熬夜到天亮…
			SETBIT CFLAG:MASTER:行動, 0
			CFLAG:MASTER:一緒に寝た = TARGET
			CFLAG:一緒に寝た = MASTER
		ELSEIF RAND:3 == 0
			PRINTFORMW %CALLNAME:TARGET%幸福地看着和孩子一起吵吵闹闹的%CALLNAME:MASTER%…
		ELSEIF RAND:2 == 0
			PRINTFORMW 和在发呆的孩子３人一起玩了簡単的游戏、就这样快快乐乐地度过了…
		ELSE
			PRINTFORM %CALLNAME:TARGET%对养育孩子而感到不安、
			IF RAND:3 == 0
				PRINT 梳着頭髪的同时
			ELSEIF RAND:2 == 0
				PRINT 温柔地抚摸着的同时
			ELSE
				IF  TALENT:娇小
					PRINT 抱着的同时
				ELSE
					PRINT 保持肌肤接觸的同时
				ENDIF
			ENDIF
			PRINTFORMW 倾听着这些烦恼…
		ENDIF
	;父親はMASTER以外（現状保留）
	;ELSEIF FATHER > MASTER
	ENDIF
;妊娠中
ELSEIF TALENT:妊娠 && COND("手一杯", MASTER) + COND("同衾イベント", MASTER) == 0 && RAND:4
	FATHER = COND("精液の優勢")
	DRAWLINE
	;妊娠しているのはMASTER
	IF TARGET == MASTER
		PRINTFORMW %CALLNAME:MASTER%由于不想出門、决定在自己的房间里休息。
		SETBIT CFLAG:MASTER:行動, 0
	;同居していない
	ELSEIF COND("同居中") == 0
		CLEARLINE 1
	;父親はMASTERor既にいないor恋慕
	ELSEIF FATHER == MASTER || FATHER < 0 || TALENT:恋慕
		PRINTFORMW %CALLNAME:MASTER%决定去看看%CALLNAME:TARGET%的情况。
		IF TALENT:叛逆 && TALENT:狂気
			IF RAND:4 == 0
				PRINTFORMW %CALLNAME:TARGET%用空洞的眼神看着隆起的肚子…
			ELSEIF RAND:3 == 0
				PRINTFORMW %CALLNAME:TARGET%毫无感情地俯视着把耳朵贴在肚子上的%CALLNAME:MASTER%…
			ELSEIF RAND:2 == 0
				PRINTFORMW %CALLNAME:TARGET%空虚地笑着和肚子里的孩子对着话…
			ELSE
				PRINTFORMW %CALLNAME:TARGET%開心地和肚子里的孩子对话、
				PRINTFORMW 但看到%CALLNAME:MASTER%时突然開始流泪了…
			ENDIF
		ELSEIF FATHER == MASTER && RAND:4
			IF RAND:4 == 0
				PRINTFORMW %CALLNAME:TARGET%睡意朦胧地出来迎接、今夜决定一起睡觉…
				SETBIT CFLAG:MASTER:行動, 0
				CFLAG:MASTER:一緒に寝た = TARGET
				CFLAG:一緒に寝た = MASTER
			ELSEIF RAND:3 == 0
				PRINTFORMW …但是、%CALLNAME:TARGET%好像已经睡着了、所以只是吻了一下便回去了…
			ELSEIF RAND:2
				PRINTFORMW %CALLNAME:TARGET%眯起眼睛開心地看着把耳朵贴在肚子上的%CALLNAME:MASTER%…
			ELSE
				PRINTFORMW 和%CALLNAME:TARGET%聊着生孩子的事情、愉快地度过了時間…
			ENDIF
		ELSE
			PRINTFORM %CALLNAME:TARGET%总是感到不安、
			IF RAND:3 == 0
				PRINT 梳着頭髪的同时
			ELSEIF RAND:2 == 0
				PRINT 温柔地抚摸着的同时
			ELSE
				IF  TALENT:娇小
					PRINT 抱着的同时
				ELSE
					PRINT 保持肌肤接觸的同时
				ENDIF
			ENDIF
			PRINTFORMW 倾听着这些烦恼…
		ENDIF
	;父親はMASTER以外（現状保留）
	;ELSEIF FATHER > MASTER
	ENDIF
ENDIF

TARGET = LOCAL

;---------------------------------------------------------
;妊娠判定格子
;シチュエーションARGSでARGが妊娠するかどうかを判定する。ARG:1は精液量。
;ARGSは"通常", "調教外", "一般人相手"
;SPERMで子宮に入る精子量、IS_FATHERで誰が父親になるかの番号管理を行う
;ARGS:1でどっちに注がれたかを判断する
;---------------------------------------------------------
@CHECK_PREGNANT, ARG, ARG:1, ARGS, ARGS:1
#DIM SPERM
#DIM SPERM_DENS
#DIM IS_FATHER

;精液量がない場合
SIF ARG:1 <= 0
	RETURN 0

;調教中かどうかとPLAYERが一般人かどうかを格子
SELECTCASE ARGS
;指定されている場合にはARGSの変化无
CASE "一般人相手", "調教外", "通常"
CASE "便利店店员"
CASE "公园厕所中不认识的男人"
CASEELSE
	IF TALENT:PLAYER:一般人
		ARGS = 一般人相手
	ELSEIF COND("調教中")
		ARGS = 通常
	ELSE
		ARGS = 調教外
	ENDIF
ENDSELECT

;SPERMで子宮に入る精子量、IS_FATHERで誰が父親になるかの番号管理を行う
SPERM = ARG:1 * (10 + RAND:21)
SPERM_DENS = 90 + RAND:21

SELECTCASE ARGS
CASE "一般人相手"
	;処女は判定しない
	SIF TALENT:ARG:処女
		RETURN 0
	IS_FATHER = -4
	;精液量は館のメンバーに比べると少ない
	SPERM = MAX(SPERM/2, 1)
;修改
CASE "便利店店员"
	IS_FATHER = -101
CASE "公园厕所中不认识的男人"
	IS_FATHER = -102
CASE "公园露出中只有一面之缘的男人"
	IS_FATHER = -103
CASE "教会的孩子"
	IS_FATHER = -104
CASE "学校的老师"
	IS_FATHER = -105
CASE "不认识的同学"
	IS_FATHER = -106
CASE "高年级的学长"
	IS_FATHER = -107
CASE "体育课的助教"
	IS_FATHER = -108
CASE "生理课老师"
	IS_FATHER = -109
CASE "做爱部的大叔"
	IS_FATHER = -110
CASE "做爱部的部员"
	IS_FATHER = -111
;----
CASEELSE
	SELECTCASE ARG
	;調教者の妊娠格子
	CASE PLAYER
		IS_FATHER = TARGET

	;TARGETやASSI等の妊娠格子
	CASEELSE
		SELECTCASE BASE:PLAYER:識別番号
		CASE IS > 0
			IS_FATHER = PLAYER
		CASEELSE
			IS_FATHER = -4
		ENDSELECT

		;絶頂を迎える事で確率上昇
		SIF NOWEX:ARG:絶頂数合計
			SPERM += MIN(NOWEX:ARG:絶頂数合計, 2)/2

		IF COND("子宮挿入フラグ", ARG)
			SPERM *= 10
		;挿入子宮口玩弄の時には妊娠率上昇
		ELSEIF SELECTCOM == 67
			TIMES SPERM, 1.20
		ENDIF
	ENDSELECT
ENDSELECT

;妊娠判定フラグ
SELECTCASE ARGS:1
CASE "肛門", "Ａ"
	BASE:ARG:Ａ胎内精液量 += SPERM * SPERM_DENS
CASEELSE
	BASE:ARG:胎内精液量 += SPERM * SPERM_DENS
ENDSELECT

;避妊中で妊娠不可能なら精子は受け入れない
SIF COND("避妊中", ARG) && COND("避妊中だが妊娠可能", ARG) == 0
	SPERM = 0
;人形だと生殖能力没有
SIF IS_FATHER > 0 && COND("種族：人形", IS_FATHER)
	SPERM = 0

SELECTCASE ARGS:1
CASE "肛門", "Ａ"
	;吸精体質の人は精子を頂いちゃう
	SIF TALENT:ARG:吸精体質
		SPERM = MAX(MIN(1 + EX:ARG:Ａ中出し*2, SPERM), SPERM/POWER(2, TALENT:ARG:吸精体質) )
	;夜袭などでの妊娠率を落とす
	SIF COND("調教中", ARG) == 0
		TIMES SPERM, 0.66

	;父親一時フラグ
	IF SPERM && BASE:ARG:胎児 == 0 && TALENT:ARG:Ａ妊娠体質
	 	IF BASE:ARG:Ａ胎内精液量 == 0 || COND("精液の優勢", ARG) == 0 || RAND:3 == 0
	 		SELECTCASE IS_FATHER
			CASE IS > 0
				BASE:ARG:精液の優勢 = BASE:IS_FATHER:識別番号
			CASEELSE
				BASE:ARG:精液の優勢 = IS_FATHER
			ENDSELECT
		ENDIF
	ENDIF

	SIF COND("精液の優勢", ARG)
		BASE:ARG:Ａ胎内精子量 += SPERM

	;排卵誘発剤
	SIF TEQUIP:ARG:排卵誘発剤 && TALENT:ARG:Ａ妊娠体質
		SETBIT BASE:ARG:バイオリズム, 13

	;中出しされた最新の日期を上書き
	CFLAG:ARG:Ａ最新中出し日 = DAY
CASEELSE
	;吸精体質の人は精子を頂いちゃう
	SIF TALENT:ARG:吸精体質
		SPERM = MAX(MIN(1 + EX:ARG:Ｖ中出し*2, SPERM), SPERM/POWER(2, TALENT:ARG:吸精体質) )
	;夜袭などでの妊娠率を落とす
	SIF COND("調教中", ARG) == 0
		TIMES SPERM, 0.66

	;父親一時フラグ
	IF SPERM && BASE:ARG:胎児 == 0 && TALENT:ARG:Ａ妊娠体質 == 0
	 	IF BASE:ARG:胎内精子量 == 0 || COND("精液の優勢", ARG) == 0 || RAND:3 == 0
	 		SELECTCASE IS_FATHER
			CASE IS > 0
				BASE:ARG:精液の優勢 = BASE:IS_FATHER:識別番号
			CASEELSE
				BASE:ARG:精液の優勢 = IS_FATHER
			ENDSELECT
		ENDIF
	ENDIF

	SIF COND("精液の優勢", ARG)
		BASE:ARG:胎内精子量 += SPERM

	;排卵誘発剤
	SIF TEQUIP:ARG:排卵誘発剤 && TALENT:ARG:Ａ妊娠体質 == 0
		SETBIT BASE:ARG:バイオリズム, 13

	CFLAG:ARG:最新中出し日 = DAY
ENDSELECT


code:
@BIRTH_RECORD, ARGS, ARG, ARG:1
#DIM LCOUNT
#DIM LCOUNT2
#DIM NEXT_LINE
#DIM DIVISION_NUM
#DIM FATHER_NUM
#DIM BOY_NUM
#DIM GIRL_NUM
#DIM FATHER_COUNT
#DIMS FATHER_NAME
#DIMS CHILD, 2

SELECTCASE ARGS
CASE "出産記録：保存"

	IF TALENT:ARG:出産経験 > 99
		CALL PRINT_STR, "イエロー_出産记录的记录总数达到了限界。不能再记录了_W"
		RETURN ARG:1
	ENDIF
	;親番号の保存
	SELECTCASE CSTR:ARG:出産記録
	CASE ""

	CASEELSE
		CSTR:ARG:出産記録 += "/"
	ENDSELECT
	SELECTCASE BASE:ARG:精液の優勢
	CASE IS > 0
		CSTR:ARG:出産記録 += @"%TOSTR(BASE:ARG:精液の優勢)%"
	CASEELSE
		CSTR:ARG:出産記録 += "不"
	ENDSELECT
	;性別の保存
	SELECTCASE ARG:1
	CASE 1
		CSTR:ARG:出産記録 += "_1"
	CASEELSE
		CSTR:ARG:出産記録 += "_0"
	ENDSELECT
	;この後の処理でも性別判定は必要なのでRETURNで性別を返す
	RETURN ARG:1

CASE "出産記録：書き出し"

	SIF CSTR:ARG:出産記録 == ""
		RETURN 0

	CALL DRAWLINES, "□ 出産の記録 □/出産の記録"

	SIF CONFIG("項目：出産の記録")
		RETURN 0

	VARSET LOCALS
	NEXT_LINE = 0

	;兒童1人ずつの情報に分離
	SPLIT CSTR:ARG:出産記録, "/", LOCALS
	;分離した数を記録
	DIVISION_NUM = RESULT

	FOR LCOUNT, 1, CHARANUM
		;自分自身は父親にならない
		SIF LCOUNT == ARG
			CONTINUE

		BOY_NUM = 0
		GIRL_NUM = 0

		FOR LCOUNT2, 0, DIVISION_NUM
			;親番号と性別を分離
			SPLIT LOCALS:LCOUNT2, "_", CHILD

			;文字列を整数に
			IF ISNUMERIC(CHILD:0)
				FATHER_NUM = TOINT(CHILD:0)
			ELSE
				FATHER_NUM = 0
			ENDIF

			SIF FATHER_NUM != BASE:LCOUNT:識別番号
				CONTINUE
			SELECTCASE TOINT(CHILD:1)
			CASE 1
				BOY_NUM += 1
			CASEELSE
				GIRL_NUM += 1
			ENDSELECT
		NEXT

		IF BOY_NUM + GIRL_NUM
			PRINTFORM [{LCOUNT,2}]%TEXT_LJ(@"和%CALLNAME:LCOUNT%的孩子", 26)%共{BOY_NUM + GIRL_NUM,2}人(
			CALL GENDER_SYMBOL, "男", 1
			PRINTFORM {BOY_NUM,2}:
			CALL GENDER_SYMBOL, "女", 1
			PRINTFORM {GIRL_NUM,2})
			NEXT_LINE += 1
			SIF NEXT_LINE%2 == 0
				PRINTL
		ENDIF
	NEXT

	;部員以外の兒童
	FOR LCOUNT, 0, 4

		BOY_NUM = 0
		GIRL_NUM = 0

		FOR LCOUNT2, 0, DIVISION_NUM
			;親番号と性別を分離
			SPLIT LOCALS:LCOUNT2, "_", CHILD

			IF (LCOUNT == 0 && CHILD:0 == "旦那") || (LCOUNT == 1 && CHILD:0 == "彼氏") || (LCOUNT == 2 && CHILD:0 == "元彼") || (LCOUNT == 3 && CHILD:0 == "不")
				SELECTCASE TOINT(CHILD:1)
				CASE 1
					BOY_NUM += 1
				CASEELSE
					GIRL_NUM += 1
				ENDSELECT
			ENDIF
		NEXT

		SELECTCASE LCOUNT
		CASE 0
			FATHER_NAME = 和丈夫的孩子
		CASE 1
			FATHER_NAME = 和男朋友的孩子
		CASE 2
			FATHER_NAME = 和前男友的孩子
		CASE 3
			FATHER_NAME = 父亲不明的孩子
		ENDSELECT

		IF BOY_NUM + GIRL_NUM
			PRINTFORM [XX]%TEXT_LJ(FATHER_NAME, 26)%{BOY_NUM + GIRL_NUM,2}人(
			CALL GENDER_SYMBOL, "男", 1
			PRINTFORM {BOY_NUM,2}:
			CALL GENDER_SYMBOL, "女", 1
			PRINTFORM {GIRL_NUM,2})
			NEXT_LINE += 1
			SIF NEXT_LINE%2 == 0
				PRINTL
		ENDIF
	NEXT

	SIF NEXT_LINE%2
		PRINTL

;兒童を生まれた順番に表示していく方式
CASE "出産記録：書き出し２"

	SIF CSTR:ARG:出産記録 == ""
		RETURN 0

	CALL DRAWLINES, "□ 出産の記録 □/出産の記録"

	SIF CONFIG("項目：出産の記録")
		RETURN 0

	VARSET LOCALS
	NEXT_LINE = 0

	;兒童1人ずつの情報に分離
	SPLIT CSTR:ARG:出産記録, "/", LOCALS
	;分離した数を記録
	DIVISION_NUM = RESULT

	FOR LCOUNT, 0, DIVISION_NUM
		;親番号と性別を分離
		SPLIT LOCALS:LCOUNT, "_", CHILD

		;文字列を整数にしてから父親の識別番号と照合
		IF ISNUMERIC(CHILD:0)
			FATHER_NUM = TOINT(CHILD:0)
			FATHER_COUNT = 0

			FOR LCOUNT2, 1, CHARANUM
				;自分自身は父親にならない
				IF LCOUNT2 == ARG
					FATHER_COUNT += 1
					CONTINUE
				ENDIF
				;部員内に父親がいる
				SELECTCASE FATHER_NUM
				CASE BASE:LCOUNT2:識別番号
					PRINTFORM [{LCOUNT + 1,2}]%TEXT_LJ(@"和%CALLNAME:LCOUNT2%的孩子(\@ TOINT(CHILD:1) ? 儿子 # 女儿 \@)", 34)%
					NEXT_LINE += 1
				CASEELSE
					FATHER_COUNT += 1
				ENDSELECT
			NEXT
			;父親が識別番号持ちだが誰の識別番号とも一致しなかった場合(角色削除でもしない限り起きないとは思うが念のため)
			IF FATHER_COUNT == (CHARANUM - 1)
				PRINTFORM [{LCOUNT + 1,2}]%TEXT_LJ(@"父親不明的小孩(\@ TOINT(CHILD:1) ? 儿子 # 女儿 \@)", 34)%
				NEXT_LINE += 1
			ENDIF
		;父親が識別番号を持たない場合(現状ではないと思うが念のため)
		ELSE
			SELECTCASE CHILD:0
			CASE "旦那"
				FATHER_NAME = 和丈夫的孩子
			CASE "彼氏"
				FATHER_NAME = 和男朋友的孩子
			CASE "元彼"
				FATHER_NAME = 和前男友的孩子
			CASEELSE
				FATHER_NAME = 父亲不明的孩子
			ENDSELECT
			PRINTFORM [{LCOUNT + 1,2}]%TEXT_LJ(@"%FATHER_NAME%(\@ TOINT(CHILD:1) ? 儿子 # 女儿 \@)", 34)%
			NEXT_LINE += 1
		ENDIF

		SIF NEXT_LINE%2 == 0
			PRINTL
	NEXT

	SIF NEXT_LINE%2
		PRINTL

ENDSELECT

;ARG:1がARGの兒童を産んだ数を返す関数
@BABY(ARG, ARG:1)
#FUNCTION
#DIM LCOUNT
#DIM DIVISION_NUM
#DIMS CHILD, 2

;ARG:1を省略でTARGET
SIF ARG:1 == 0 && TARGET
	ARG:1 = TARGET
SIF ARG < 0 || ARG:1 < 0
	RETURNF 0
SIF ARG == ARG:1
	RETURNF 0
SIF CSTR:(ARG:1):出産記録 == ""
	RETURNF 0

VARSET LOCALS

;兒童1人ずつの情報に分離
SPLIT CSTR:(ARG:1):出産記録, "/", LOCALS
;分離した数を記録
DIVISION_NUM = RESULT
LOCAL = 0

FOR LCOUNT, 0, DIVISION_NUM
	;親番号と性別を分離
	SPLIT LOCALS:LCOUNT, "_", CHILD

	SIF TOINT(CHILD:0) == BASE:ARG:識別番号
		LOCAL += 1
NEXT

RETURNF LOCAL

;性別記号を出力する関数
@GENDER_SYMBOL, ARGS, ARG
GETFONT
CHKFONT "Times New Roman"
SIF RESULT && ARG == 1
	SETFONT "Times New Roman"
SELECTCASE ARGS
CASE "男"
	SETCOLOR DEF_COLOR("青")
	PRINTFORM %UNI("男性")%
CASE "女"
	SETCOLOR DEF_COLOR("赤")
	PRINTFORM %UNI("女性")%
ENDSELECT
SETFONT RESULTS
RESETCOLOR


code:
;-------------------------------------------------
;休憩したときに起きるイベント
;-------------------------------------------------
@EVENT_REST_PRE
SIF ASSI <= 0
	RETURN 0
SIF BASE:MASTER:精力 >= MAXBASE:MASTER:精力
	RETURN 0

SWAP ASSI, TARGET
CALL EVENT_REST
SWAP ASSI, TARGET

@EVENT_REST
SIF ABL:奉仕精神 < RAND:10 || RAND:3
	RETURN 0
SIF TALENT:恋慕 == 0 && TALENT:淫乱 == 0 && MARK:屈服刻印 < 3
	RETURN 0
;とりあえずは性別の制限つきで
SIF TALENT:男性 || PENIS(MASTER) == 0
	RETURN 0

CALL CALC, "随机選択"

;お浴室
;SIF FIRSTCOM(94)
	CALL CALC, "随机選択", 1
;掏耳朵
SIF ABL:奉仕精神 >= 3 && COND("指の使用")
	CALL CALC, "随机選択", 2

;エッチという名の助手の報酬(とりあえずは主人が男性で相手が女のときのみ)
;IF TALENT:MASTER:男性 && TALENT:男性 == 0
;	SIF TALENT:好色 + TALENT:淫乱 || (COND("欲求不満") && ABL:欲望 >= 4)
;		CALL CALC, "随机選択", 3
;ENDIF

;添い寝
;CALL CALC, "随机選択", 2
;茶
;CALL CALC, "随机選択", 3
;マッサージ
;SIF COND("指の使用") && ABL:指靈活 >= 3
;	CALL CALC, "随机選択", 4

SELECTCASE RESULT
CASE 1
	LOCALS = 休憩：お浴室
CASE 2
	LOCALS = 休憩：掏耳朵
CASE 3
	LOCALS = 休憩：掃除

;CASE
;	LOCALS = 休憩：エッチ
;CASE
;	LOCALS = 休憩：添い寝
;CASE
;	LOCALS = 休憩：茶
;CASE
;	LOCALS = 休憩：マッサージ

CASEELSE
	RETURN 0
ENDSELECT

DRAWLINE

;LOCALで得た経験を管理する予定
VARSET LOCAL

SELECTCASE LOCALS
CASE "休憩：掃除"

CASE "休憩：お浴室"
	PRINTFORMW %CALLNAME:MASTER%在%TEXTS("現在位置")%里刚一%TEXTR("休憩/躺下/放松")%\@ TARGET == ASSI ? 、%TEXTS("助手の名称")%的 # \@%CALLNAME:TARGET%就来了
	PRINTFORM 问了一下有什么事……原来是、
	IF RAND:2
		PRINTFORMW 想在浴室里帮忙搓搓背
	ELSE
		PRINTFORMW 想在浴室里帮忙洗身子
	ENDIF
	PRINTFORMW 当然、肯定不止字面上的意思…

	PRINTL
	PRINTFORML ……那么要怎么做呢？
	PRINTFORML  [ 0] 接受请求
	PRINTFORML [100] 委婉地拒绝
	CALL INPUT_SELECT, 1, 100

	SELECTCASE RESULT
	CASE 100
		;「あ……そ、そう。 …休息、今度は付き合ってよね？」
		PRINTFORMW %CALLNAME:TARGET%对%CALLNAME:MASTER%没什么兴致的様子有点失落、老实地离開了…
		RETURN 0
	CASE 0

		PRINTL
		PRINTFORML －浴場－
		PRINTFORMW %CALLNAME:MASTER%按照指示在浴室的垫子上躺下、放松了全身的力气、享受着%CALLNAME:TARGET%的本领
		PRINTFORM 用充满泡沫的%TEXTS("胸部")%轻蹭着擦洗你全身的%CALLNAME:TARGET%、
		IF ABL:技巧 >= ABL:MASTER:技巧
			PRINTFORM 慢慢地变大了
		ELSE
			PRINTFORM 变得有些硬了
		ENDIF
		PRINTFORM %TEXTR("阴茎/比较脏的部分")%
		IF COND("Ｖ熟達") && TALENT:MASTER:童貞 == 0
			SIF ABL:技巧 < ABL:MASTER:技巧
				PRINTFORM 用%TEXTR("指/胸/口/素股/接吻")%刺激着使你勃起後
			IF ABL:Ｖ拡張 > SIZE("阴茎", MASTER)
				PRINTFORM 直到根部
			ELSEIF ABL:Ｖ拡張 == SIZE("阴茎", MASTER)
				PRINTFORM 认真地
			ELSE
				PRINTFORM 一直到最里面
			ENDIF
			PRINTFORM 阴道内%TEXTR("吸入了/吞入了")%、
			IF TALENT:感情缺乏 + TALENT:冷静
				PRINTFORM 一边报告着%CALLNAME:MASTER%的反応
			ELSEIF TALENT:開朗 && RAND:3
				PRINTFORM 用荡漾着快楽的笑顔看着%CALLNAME:MASTER%的脸
			ELSEIF TALENT:強気 + TALENT:高慢 && RAND:3
				PRINTFORM 一边用很有余裕的轻松的笑顔看着你
			ELSEIF TALENT:巨乳 && RAND:3
				IF RAND:3 == 0 && TALENT:巨乳 >= 2
					IF RAND:2
						PRINTFORM 一边夸张地揺着豊満的胸
					ELSE
						PRINTFORM 一边悠悠地揺着看起来很重的胸部
					ENDIF
				ELSEIF RAND:3 == 0
					PRINTFORM 一边揺着大大的胸部
				ELSEIF RAND:2
					PRINTFORM 一边给你展示着揺动的胸部
				ELSE
					PRINTFORM 一边揉着胸
				ENDIF
			ELSE
				PRINTFORM 身体害羞地泛红
			ENDIF
			IF ABL:腰靈活 >= 5 && RAND:2
				PRINTFORMW 展示自己淫靡地动腰姿势…
			ELSEIF ABL:Ｖ感覚 >= MAX(3, ABL:MASTER:技巧) && RAND:2
				PRINTFORMW 可愛地喘息着…
			ELSEIF RAND:2
				PRINTFORMW 激烈地刺激着…
			ELSE
				PRINTFORMW 动着腰…
			ENDIF
			LOCALS = 休憩：お浴室で本番
		ELSE
			IF TALENT:巨乳
				PRINTFORM 用自豪的胸部夹着
			ELSEIF ABL:舌靈活 >= MAX(ABL:舌靈活, ABL:指靈活, 1)
				PRINTFORM 全部含入口中
			ELSEIF ABL:指靈活 >= MAX(ABL:舌靈活, 1)
				PRINTFORM 动起了手指
			ENDIF
			PRINTFORMW 時而激烈、時而柔緩地刺激着…
		ENDIF

		CALL KOJO_EVENT, LOCALS

		PRINTL
		PRINTFORMW %CALLNAME:TARGET%让%CALLNAME:MASTER%彻底舒服了一遍、之后普通地把身体清洁了一遍、期待着下次的休息…
	ENDSELECT

	;PRINTFORM マットに横たわった%CALLNAME:TARGET%は%CALLNAME:MASTER%に向かって股を開き、洗いにくい体の奥の部分を洗ってくれるようにおねだりしてきました

CASE "休憩：掏耳朵"
	PRINTFORMW %CALLNAME:MASTER%在%TEXTS("現在位置")%里刚一%TEXTR("休憩/躺下/放松")%\@ TARGET == ASSI ? 、%TEXTS("助手の名称")%的 # \@%CALLNAME:TARGET%就来了
	PRINTFORM 问了一下有什么事……原来是、
	IF COND("種族：拉米亚")
		PRINTFORMW 想要帮忙掏耳朵
		PRINTFORMW 占据了床上大部分面积的%CALLNAME:TARGET%将身体盘成平缓的斜坡状、招手示意让%CALLNAME:MASTER%平躺在这里…
	ELSE
		PRINTFORMW 想要用膝枕帮忙掏耳朵
		PRINTFORMW 座在床上的%CALLNAME:TARGET%哒哒地轻敲着膝盖、示意让%CALLNAME:MASTER%躺过来…
	ENDIF

	PRINTL
	PRINTFORML ……那么要怎么做呢？
	PRINTFORML  [ 0] 乖乖地享受耳掃除
	PRINTFORML  [ 1] 耳掃除的同时…做些恶作剧？
	PRINTFORML [100] 委婉地拒绝
	CALL INPUT_SELECT, 2, 100

	SELECTCASE RESULT
	CASE 100
		;「あ……そ、そう。 …休息、今度は付き合ってよね？」
		PRINTFORMW %CALLNAME:TARGET%对%CALLNAME:MASTER%没什么兴致的様子有点失落、老实地离開了…
		RETURN 0
	CASE 0
		PRINTFORMW %CALLNAME:MASTER%被用膝枕\@ COND("種族：拉米亚") ? （？） # \@的姿势掻着耳朵、迷迷糊糊地享受了短暂的休息时光
		IF ABL:技巧 >= 5
			PRINTFORMW %CALLNAME:TARGET%用完全不会造成疼痛感的灵巧手法给両耳做了掃除、用温柔的手梳理着%CALLNAME:MASTER%的頭髪…
		ELSE
			IF RAND:10 > ABL:技巧
				PRINTFORMW 虽然偶尔会因为轻微的疼痛而惊醒、不过这技术很快就会进步的吧…
			ELSE
				PRINTFORMW %CALLNAME:TARGET%%TEXTR("因取下了大块的耳垢而開心着…/因为没有取下耳垢、好像有点遗憾…/表情认真地清理着耳朵、看起来很開心…")%
			ENDIF
		ENDIF
	CASE 1
		PRINTFORMW %CALLNAME:MASTER%被用膝枕\@ COND("種族：拉米亚") ? （？） # \@的姿势掻着耳朵的同时、决定在这期间做一些恶作剧来更加愉悦地度过这时间
		IF MIN(RAND:10, 5) > ABL:技巧 && RAND:2
			PRINTFORM …欸、得意忘形的
			IF COND("種族：拉米亚") == 0 && RAND:3
				PRINTFORM %TEXTR("足裏/腿内侧")%
			ELSE
				PRINTFORM %TEXTS("腋")%下
			ENDIF
			PRINTFORMW 挠痒痒大失敗！
			PRINTFORM %CALLNAME:TARGET%
			IF TALENT:感情缺乏 + TALENT:冷静
				PRINTFORM 『…自作自受』
			ELSEIF TALENT:強気 + TALENT:高慢
				PRINTFORM 『搞砸了…』
			ELSE
				PRINTFORM 『怎、怎么办…』
			ENDIF
			PRINTFORML 的表情僵硬住了、看着%CALLNAME:MASTER%用敲打反击了回去…
			PRINTFORMW ※深度掃除的时候不要挠痒哟※
		ELSE
			PRINTFORM %CALLNAME:TARGET%
			IF TALENT:感情缺乏 + TALENT:冷静 && RAND:3
				PRINTFORMW 以毫无反应的状态继续默默地挖着耳朵…
				PRINTFORMW 就在%TEXTR("乳頭被恶作剧/淫核被剥弄/膣口被手指进入")%时、吓得身体一跳之後～惊讶地看了一眼%CALLNAME:MASTER%
				PRINTFORMW 似乎没有生气的样子、%CALLNAME:TARGET%改变姿势接受了%CALLNAME:MASTER%的恶作剧、每次敏感的部分被撫摸的时候都会漏出甜美的声音、继续掏着耳朵…
			ELSEIF TALENT:開朗 && RAND:3
				PRINTFORMW 在每次被%CALLNAME:MASTER%%TEXTR("撫摸阴部/触摸/搓揉胸部")%的时候都会发出舒服的声音、有时发出『nia~』的声音、兴致勃勃地掏着耳朵…
			ELSEIF TALENT:強気 + TALENT:高慢 && RAND:3
				PRINTFORMW 虽然在恶作剧開始的时候抱怨了几句、但在意识到这是自己魅力的体现后、意外地展现出了寛容的態度
				PRINTFORMW …欸、在抵抗弱的期间、试着指出%TEXTR("胖次已经湿了/阴蒂变大了/乳頭硬邦邦的")%这件事的时候、被说『不要得意忘形』地敲打教训了…
			ELSEIF TALENT:巨乳 && RAND:3
				PRINTFORMW 压住%CALLNAME:MASTER%的脸的沉重的乳房被找到机会反撃了、享受着一边被恶作剧一边掏耳朵的美好时光…
			ELSE
				PRINTFORMW 用完全不会造成疼痛感的灵巧手法给両耳做着掃除、对%CALLNAME:MASTER%的色情恶作剧也完全不生气、用温柔的手法梳理着%CALLNAME:MASTER%的頭发…
			ENDIF
		ENDIF
	ENDSELECT
ENDSELECT


code:
;-------------------------------------------------
;性処理関連が複雑化してきたのでまとめる
;-------------------------------------------------
;@CANTRAINの条件付きver
;ARGSにはTRAINNAMEが入る。性欲解消(你がpartnerの性処理に回る)の際にはTRAINNAMEの前に"性欲解消"と付ける。例)ARGS = 性欲解消正常位
;具現については[本人が具現の使い手]＆[兩面通吃]の場合に限り性処理でも使用可としていたがバグに見えるっぽいのでオミット
;-------------------------------------------------
@CANSS(ARGS, ARG)
#FUNCTION
#DIM IS_REV
;竿役
#DIM ACTOR_USEP

SIF ARG < 0 || ARGS == ""
	RETURNF 0
SIF ARG == 0 && TARGET > 0
	ARG = TARGET

ACTOR_USEP = 0
IS_REV = 0

;性欲解消？
IF STRCOUNT(ARGS, "性欲解消")
	ARGS = %SUBSTRINGU(ARGS, 4, -1)%
	IS_REV = 1
ENDIF

;CANTRAINを満たしている事が前提なので処女の場合は無視して良い
SIF CANTRAIN(ARGS, ARG) == 0
	RETURNF 0

;性別条件などがあるなら見る。性欲解消は該当箇所の感覚３ＬＶが目安
SELECTCASE ARGS
CASE "舐陰"
	SIF TALENT:ARG:男性
		RETURNF 0
	IF IS_REV
		SIF ABL:ARG:Ｃ感覚 < 3 || TALENT:ARG:Ｃ性向 + TALENT:ARG:淫核 + COND("陰核肥大", ARG) == 0
			RETURNF 0
	ENDIF
CASE "手淫", "乳交"
	ACTOR_USEP = PLAYER
CASE "口交"
	ACTOR_USEP = PLAYER
	IF IS_REV
		SIF ABL:ARG:精液中毒 < 3 || COND("精飲弱点", ARG) + TALENT:ARG:气味弱点 == 0
			RETURNF 0
	ENDIF
CASE "正常位", "後背位", "屈曲位", "対面座位", "背面座位", "騎乗位"
	ACTOR_USEP = PLAYER
	SIF TALENT:ARG:男性
		RETURNF 0
	SIF ABL:ARG:Ｖ拡張 < SIZE("阴茎", PLAYER)
		RETURNF 0
	IF IS_REV
		SIF ABL:ARG:Ｖ感覚 < 3 || COND("Ｖ熟達", ARG) == 0
			RETURNF 0
	ENDIF
CASE "肛門正常位", "肛門後背位", "肛門屈曲位", "肛門対面座位", "肛門背面座位", "肛門騎乗位"
	ACTOR_USEP = PLAYER
	SIF ABL:ARG:Ａ拡張 < SIZE("阴茎", PLAYER)
		RETURNF 0
	IF IS_REV
		SIF ABL:ARG:Ａ感覚 < 3 || COND("Ａ熟達", ARG) == 0 || TALENT:ARG:Ａ性向 + TALENT:ARG:尻穴狂い + TALENT:ARG:男性 == 0
			RETURNF 0
	ENDIF
CASE "尿道正常位", "尿道後背位", "尿道屈曲位", "尿道対面座位", "尿道背面座位", "尿道騎乗位"
	ACTOR_USEP = PLAYER
	SIF ABL:ARG:尿道拡張 < SIZE("阴茎", PLAYER)
		RETURNF 0
	IF IS_REV
		SIF ABL:ARG:尿道感覚 < 3 || COND("尿道熟達", ARG) == 0 || TALENT:ARG:尿道性感 + TALENT:ARG:尿道狂 == 0
			RETURNF 0
	ENDIF
CASE "乳頭奸"
	ACTOR_USEP = PLAYER
	SIF COND("Ｂ熟達", ARG) == 0
		RETURNF 0
CASE "奉仕舐陰"
	SIF TALENT:PLAYER:男性
		RETURNF 0
CASE "舐陰莖"
	ACTOR_USEP = ARG
	IF IS_REV
		SIF ABL:ARG:Ｃ感覚 < 3 || TALENT:ARG:Ｃ性向 + TALENT:ARG:淫核 + TALENT:ARG:男性 == 0
			RETURNF 0
	ENDIF
CASE "正常位奉仕", "後背位奉仕", "騎乗位侵犯"
	ACTOR_USEP = ARG
	SIF TALENT:PLAYER:男性
		RETURNF 0
	IF IS_REV
		SIF TALENT:ARG:Ｃ性向 + TALENT:ARG:男性 == 0
			RETURNF 0
	ELSE
		SIF COND("Ｖ熟達", PLAYER) == 0
			RETURNF 0
	ENDIF
CASE "Ａ正常位奉仕", "Ａ後背位奉仕", "Ａ騎乗位侵犯"
	ACTOR_USEP = ARG
	IF IS_REV
		SIF TALENT:ARG:Ｃ性向 + TALENT:ARG:男性 == 0
			RETURNF 0
	ELSE
		SIF COND("Ａ熟達", PLAYER) == 0
			RETURNF 0
	ENDIF

;それ以外は性処理指令ではないのでダメ
CASEELSE
	RETURNF 0
ENDSELECT

;要竿役なら条件を満たすか格子
IF ACTOR_USEP
	;SIF PENIS(ACTOR_USEP) || (TALENT:ACTOR_USEP:具現 && TALENT:ACTOR_USEP:兩面通吃)
	SIF PENIS(ACTOR_USEP)
		RETURNF 1
	RETURNF 0
ENDIF

RETURNF 1